
1 About this package

The RDS package is meant to help with complete searches for relative difference sets in non-abelian groups.
Of course, it also works for abelian groups, but no special features are implemented for this case. In particular,
there is no support for multipliers.

RDS has no undocumented functions. While this is generally regarded as a feature, it leads to a quite long
manual and a lot of documentation not needed for everyday work. To make reading easier, all but the basic
chapters contain a small introductory paragraph pointing out which functions may be interesting for the
user and which are merely helper functions called by other functions.

The structure of this manual is a follows: First, there is a chapter about brute force methods which are easy
to use but are not suitable for very difficult calculations.

Then, chapter RDS:A basic example shows the use of the more advanced methods in RDS and explains the
basic idea of a complete search for difference sets with this package. After reading this chapter, you should
be able to use RDS even for large examples.

The following chapters RDS:General concepts and RDS:Invariants for Difference Sets contain the
documentation of the functions used in a search for difference sets. They explain the concepts and low level
functions which provide a lot of control over the searching process. If you are searching for difference sets in
several groups of the same order, you may find this helpful.

The next chapter shows an example of calculating a relative difference set using low level functions.

Chapter RDS:Ordered Signatures introduces another invariant for difference sets. The functions for calcu-
lating this invariant do only work effectively in a few cases, so this part of RDS is a little bit experimental.
However, the invariant is very powerful, so this chapter is kept.

In RDS:Block Designs and Projective Planes, the methods for generating a BlockDesign in the sense
of DESIGN [Soi06a] from a difference set are described. A few functions for analyzing projective planes are
given as well.

The final chapter describes a few functions which are not related to difference sets and may be useful in
other situations.

1.1 Acknowledgements

I would like to thank U. Dempwolff for supervising the thesis out of which RDS grew, and L. Soicher for
many suggestions which greatly improved the usability of this package.

1.2 Installation

RDS depends on Leonard Soicher’s DESIGN [Soi06a] package which, in turn, depends on GRAPE [Soi06b].
You need to install these packages before you can run RDS.

1. Download the package archive rdsver .ext where ver is some version number and ext is an extension
like tar.bz2, tar.gz, -win.zip or zoo.

2. Copy the archive to the directory where the other packages live. This is either the directory pkg in
the GAP root path or a local directory in your home directory (on most unix-like systems, this will
probably be ~/gap/pkg/).

3. change directory to your package directory and unpack the archive by using the right one of the following
commands:

tar -xjf rdsver .tar.bz2

tar -xzf rdsver .tar.gz

zoo -extract rdsver .zoo

unzip rdsver -win.zip

(replace ver with the version number)

4. start GAP. If you have unpacked the archive to ’gap/pkg’ in your home directory, you might have to
use ”gap -l ’homedir/gap;’ ” where homedir is the path of your home directory (use ’pwd’ to find out
what it is, if you don’t know it)

5. Type LoadPackage("rds"); to load RDS

For a test, see the examples in chapters RDS:AllDiffsets and OneDiffset and RDS:A basic example.

1.3 Verbosity

There are two info classes that control the about of additional information RDS prints:

1 I InfoRDS V

Some methods of the RDS package print additional information if InfoRDS is set to a level of 1 or higher.
At level 0, no information is output. The default value is 1.

2 I DebugRDS V

Some methods of the RDS package print additional information if DebugRDS is set to a level of 1 or higher.
At level 0, no information is output. The default level is 0. Expect a lot of output at level 2.

1.4 Definitions and Objects

This section lists the definition of ordinary and relative difference sets as well as the concept of partial differ-
ence sets and their development. This will be repeated in RDS:Introduction where a notion of equivalence
is introduced and the implementation in RDS is discussed.

Let G be a finite group and N ⊆ G . The set R ⊆ G with |R| = k is called a “relative difference set of order
k − λ relative to the forbidden set N ” if the following properties hold:

(a) The multiset {a · b−1: a, b ∈ R} contains every nontrivial ( 6= 1) element of G −N exactly λ times.

(b) {a · b−1: a, b ∈ R} does not contain any non-trivial element of N .

Let D ⊆ G be a difference set, then the incidence structure with points G and blocks {Dg | g ∈ G} is called
the development of D . In short: devD . Obviously, G acts on devD by multiplication from the right.

Relative difference sets with N = 1 are called (ordinary) difference sets. The development of a difference set
with N = 1 and λ = 1 is projective plane of order k − 1.

In group ring notation a relative difference set satisfies

RR−1 = k + λ(G −N )·

The set D ⊆ G is called partial relative difference set with forbidden set N , if

DD−1 = κ +
∑

g∈G−N

vgg

holds for some 1 ≤ κ ≤ k and 0 ≤ vg ≤ λ for all g ∈ G −N . If D is a relative difference set then ,obviously,
D is also a partial relative difference set.

IMPORTANT NOTE

RDS implicitly assumes that the every partial difference set contains the identity element (see the notion
of equivalence in RDS:Introduction for the mathematical reason). However, the identity must not be
contained in the lists representing partial relative difference sets.

So in RDS, the difference set [ (), (1,2,3,4,5,6,7), (1,4,7,3,6,2,5) ] is represented by the list [
(1,2,3,4,5,6,7), (1,4,7,3,6,2,5) ]. And no set of three non-trivial permutations will be accepted as
an ordinary difference set of Group((1,2,3,4,5,6,7)).

For this reason the lists returned by functions like RDS:AllDiffsets do only contain non-trivial elements
and look too short.

AllDiffsets and

OneDiffset

This chapter contains a number of examples as a very quick introduction to a few brute-force methods which
can be used to find all (or just one) relative difference sets in a small group. Full documentation of these
functions including all parameters can be found in section RDS:Brute force methods.

Do not expect too much from these methods alone! If you want to find examples of relative difference sets
in larger groups, you should familiarize with the notion of coset signatures by also reading the next chapter.

The functions RDS:AllDiffsets and RDS:OneDiffset present the easiest way to calculate relative difference
sets.

For a quick start, try this:

gap> LoadPackage("rds");;
gap> G:=CyclicGroup(7);
<pc group of size 7 with 1 generators>
gap> AllDiffsets(G);
[ [ f1, f1^3 ], [ f1, f1^5 ], [ f1^2, f1^3 ], [ f1^2, f1^6 ], [ f1^4, f1^5 ],
[ f1^4, f1^6 ] ]

gap> OneDiffset(G);
[ f1, f1^3 ]

The first is the set of all ordinary difference sets of order 2 in the cyclic group of order 7. Ok, they look too
small (recall that the order of a difference set is the number k of elements it contains minus the multiplicity
λ). Here is the reason:

Without loss of generality, every difference set contains the identity element of the group it lives in. RDS
knows this and assumes it implicitly. So difference sets of length n are represented by lists of length n − 1.

We can calculate all ordinary difference sets in G which contain the last element using RDS:AllDiffsetsNoSort.
Observe, that RDS:AllDiffsets calculates partial difference sets by adding elements to the given list which
are lexicographically larger than the last one of this list:

gap> AllDiffsetsNoSort([Set(G)[7]],G);
[ [ f1^6, f1^2 ], [ f1^6, f1^4 ] ]
gap> AllDiffsets([Set(G)[7]],G);
[ ]

You can also generate relative difference sets. Here we must give a partial difference set to start with (the
empty list is ok) and a forbidden set. Notice that a forbidden subgroup cannot be input as a group. It has
to be converted to a set.

gap> G:=ElementaryAbelianGroup(81);
<pc group of size 81 with 4 generators>
gap> N:=Subgroup(G,GeneratorsOfGroup(G){[1,2]});
Group([ f1, f2 ])
gap> OneDiffset([],Set(N),G);
[ f3, f4, f1*f3^2, f2*f3*f4, f1^2*f4^2, f2*f3^2*f4^2, f1*f2^2*f3^2*f4,
f1^2*f2^2*f3*f4^2 ]

If the parameter λ is not given, it is set to 1. Of course, we can also find difference sets with λ > 1. Here is
a (12, 2, 12, 6) difference set in SL(2, 3):

gap> G:=SmallGroup(24,3);
<pc group of size 24 with 4 generators>
gap> N:=First(NormalSubgroups(G),i->Size(i)=2);
Group([ f4 ])
gap> OneDiffset([],Set(N),G,6);
[ f1, f2, f3, f1^2, f1*f2, f1*f3, f2*f3, f1*f2*f3, f1^2*f2*f4, f1^2*f3*f4,
f1^2*f2*f3*f4 ]

To test if a set is a relative difference set, RDS:IsDiffset can be used:

gap> a:=(1,2,3,4,5,6,7);
(1,2,3,4,5,6,7)
gap> IsDiffset([a,a^3],Group(a)); #an ordinary difference set
true
gap> IsDiffset([a,a^2,a^4],Group(a)); #no ordinary difference set
false
gap> IsDiffset([a,a^2,a^4],Group(a),2); #diffset with <lambda>=2
true

In some cases, RDS:AllDiffsets and RDS:OneDiffset will refuse to work. A solution for this is to calculate
IsomorphismPermGroup for your group and then work with the image under this isomorphism.

See RDS:Brute force methods for details.



3 A basic example

This chapter shows a basic example of how to use RDS. Some of the functions used here make choices which
might not be optimal but should suffice for most “everyday” situations. If you plan to do more involved
computations, you should also see the other chapters to learn about the concepts behind these high-level
functions.

Here we will construct relative difference sets of Dembowski-Piper type “b” and order 9 as an example. We
will take the elementary abelian group as an example. The general idea is as follows: Find a “nice” normal
subgroup U and generate relative difference sets coset by coset. The normal subgroup has to be chosen such
that we know how many elements to choose from each coset modulo U .

The calculations here are very easy, a more demanding example can be found in chapter RDS:An Example
Program.

3.1 First Step: Integers instead of group elements

Difference sets are represented by lists of integers. Every difference set is assumed to contain 1. This is
assumed implicitly. So the lists representing difference sets must not contain 1 (a partial difference set of
length n is hence represented by a list of length n − 1). If a partial difference set contains 1, many functions
will produce errors.

To find Difference sets in a group, say G , begin with generating the group (and forbidden subgroup) and
defining the parameters. Like this:

gap> LoadPackage("rds");
----------------------------------------------------------------
Loading RDS 1.2
by Marc Roeder (marc_roeder@web.de)
----------------------------------------------------------------
true
gap> k:=9;;lambda:=1;;groupOrder:=81;;
gap> forbiddenGroupOrder:=9;;
gap> G:=ElementaryAbelianGroup(groupOrder);
<pc group of size 81 with 4 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> N:=Subgroup(G,GeneratorsOfGroup(G){[1,2]});
Group([ f1, f2 ])
gap> Size(N)=forbiddenGroupOrder; #just a test...
true

Once we have calculated Gdata, this will be used very often to represent the group G as it contains much
more information.

3.2 Signatures: An important tool

The “signature” of a subset S ⊆ G of a group relative to a normal subgroup U is the multiset of numbers of
elements S contains from each coset modulo U . Possible values of these numbers can be calculated a priori
for relative difference sets.

gap> sigdat:=SignatureData(Gdata,N,k,lambda,10^5);;

The argument 105 depends on your degree of impatience. Larger numbers take more time in this step, but
give better results for later reduction steps.

Now we will look for a “nice” normal subgroup. A normal subgroup is “nice”, if it has only few signatures
and the number of different entries in each signature is low. If you have different choices here do some
experiments, to see what works. Let’s see what we have:

gap> NormalSgsHavingAtMostNSigs(sigdat,1,[1..7]);
[ rec( sigs := [ [ 3, 3, 3 ] ], subgroup := Group([ f1, f2, f3 ]) ),
rec( sigs := [ [ 3, 3, 3 ] ], subgroup := Group([ f1, f2, f4 ]) ),
rec( sigs := [ [ 3, 3, 3 ] ], subgroup := Group([ f1, f2, f3*f4 ]) ),
rec( sigs := [ [ 3, 3, 3 ] ], subgroup := Group([ f1, f2, f3*f4^2 ]) ) ]

The second parameter of RDS:NormalSgsHavingAtMostNSigs is the maximal number of signatures the
subgroup may have. The third parameter gives the desired lengths of the signatures (the index of the
normal subgroup).

So in this example we have no real choice. Let’s take the first group for U . The signature means that we
have to get 3 elements from each coset modulo U . So we generate startsets of length 2 in the trivial coset
U (representing partial relative difference sets of length 3). This could be done using RDS:AllDiffsets, of
course. But here we will use another method. The function RDS:StartsetsInCoset generates startsets in U
by generating an initial set of startsets and then raising the length of each startset by 1. Then a reduction
using signatures and automorphism is performed. This is done until all startsets have the desired length or
no startset remains (in which case there is no relative difference set). For the reduction, a suitable set of
automorphisms must be chosen. This is done by the function RDS:SuitableAutomorphismsForReduction:

gap> U:=last[1].subgroup;
Group([ f1, f2, f3 ])
gap> auts:=SuitableAutomorphismsForReduction(Gdata,U);
[ <permutation group of size 303264 with 8 generators> ]
gap> startsets:=StartsetsInCoset([],U,N,2,auts,sigdat,Gdata,lambda);
#I Size 18
#I 1/ 0 @ 0:00:00.071
#I Size 8
#I 1/ 0 @ 0:00:00.038
#I -->1 @ 0:00:00.042
[ [ 4, 22 ] ]

For larger examples, this takes a wile. Taking 106 (or even more) for the generation of sigdat can save some
time here. A few remarks about the parameters of RDS:StartsetsInCoset. The first parameter [] is the
set of startsets which we start with (as we just started, this is empty). The second parameter is the coset we
use to generate startsets and third parameter is the forbidden subgroup. The fourth parameter is the length
of the startsets we want to generate (remember that 1 is assumed to be in every startset without being
listed. So we want startsets of size 3 represented by lists of length 2. Hence the 2 in this place). Instead of
auts a suitable list of groups of automorphisms of G in permutation representation may be inserted. These
are used for the reduction of startsets. For large groups auts[1] it is a good idea to add some subgroups of
auts[1] to the list (ascending in order) auts, as the reduction is done using the first group in the list and
then reducing the already reduced list again using the next group.

3.3 Change of coset vs. brute force

Now we have startsets of length 2 in U and there are two possibilities:

(1) Find 3 more elements from another coset like this:

gap> cosets:=RightCosets(G,U);
[ RightCoset(Group( [ f1, f2, f3 ] ),<identity> of ...),
RightCoset(Group( [ f1, f2, f3 ] ),f4),
RightCoset(Group( [ f1, f2, f3 ] ),f4^2) ]

gap> startsets:=StartsetsInCoset(startsets,cosets[2],N,5,auts,sigdat,Gdata,lambda);
#I Size 27
#I 1/ 0 @ 0:00:00.127
#I Size 11
#I 1/ 0 @ 0:00:00.058
#I -->1 @ 0:00:03.311
#I Size 2
#I 2/ 2 @ 0:00:00.015
#I -->2 @ 0:00:00.015
[ [ 4, 22, 5, 28, 73 ], [ 4, 22, 5, 28, 77 ] ]

And 3 more from the last one (of course, we could also change to force, but it seems to work this way. . . ).

gap> startsets:=StartsetsInCoset(startsets,cosets[3],N,8,auts,sigdat,Gdata,lambda);
#I Size 9
#I 1/ 0 @ 0:00:00.056
#I Size 1
#I 1/ 1 @ 0:00:00.006
#I -->1 @ 0:00:00.009
#I Size 1
#I 1/ 1 @ 0:00:00.006
#I -->1 @ 0:00:00.006
[ [ 4, 22, 5, 28, 73, 37, 66, 78 ] ]

So we found one difference set of order 9 in the elementary abelian group of order 81. To get the difference
set containing 1 explicitly and as a subset of G , say

gap> PermList2GroupList(Concatenation(startsets[1],[1]),Gdata);
[ f3, f1*f3^2, f4, f2*f3*f4, f1*f2^2*f3^2*f4, f1^2*f4^2, f2*f3^2*f4^2,
f1^2*f2^2*f3*f4^2, <identity> of ... ]

(2) Do a brute force search. Here we have to convert the forbidden group N into a list of integers Np. And
we have to raise the length of the startsets by one before we can start. This is due to the ordering we chose
(which is not necessarily compatible with the cosets modulo U ).

gap> Np:=GroupList2PermList(Set(N),Gdata);
[ 1, 2, 3, 6, 7, 10, 16, 19, 32 ]
gap> startsets:=ExtendedStartsetsNoSort(startsets,[1..groupOrder],Np,8,Gdata,lambda);;
gap> Size(startsets);
54
gap> foundsets:=[];;
gap> for set in startsets
> do
> Append(foundsets,AllDiffsets(set,[1..groupOrder],k-1,Np,Gdata,lambda));
> od;
gap> Size(foundsets);

162

Now foundsets contains 162 relative (9, 9, 9, 1)-difference sets (represented by lists of length 8). These are
all equivalent (as seen above). Equivalence can be tested like this:

gap> ReducedStartsets(foundsets,[Gdata.Aac],i->true,Gdata);
#I Size 162
#I 1/ 0 @ 0:00:00.001
[ [ 4, 22, 36, 39, 49, 50, 60, 61 ] ]



4 General concepts

In this chapter, we first give a definition of relative difference sets and outline a part of the theory. Then we
have a quick look at the way difference sets are represented in RDS.

After that, some basic methods for the generation of difference sets are explained.

If you already read chapter RDS:A basic example and want to know what RDS:StartsetsInCoset really
does, you may want to read this chapter. The most important method here is RDS:PermutationRepForDiffsetCalculations
as this is the function all searches start with. The main high-level function for difference set generation in
this chapter is RDS:ExtendedStartsets.

4.1 Introduction

Let G be a finite group and N ⊆ G . The set R ⊆ G with |R| = k is called a “relative difference set of order
k − λ relative to the forbidden set N ” if the following properties hold:

(a) The multiset {a · b−1: a, b ∈ R} contains every nontrivial ( 6= 1) element of G −N exactly λ times.

(b) {a · b−1: a, b ∈ R} does not contain any non-trivial element of N .

Relative difference sets with N = 1 are called (ordinary) difference sets. As a special case, difference sets
with N = 1 and λ = 1 correspond to projective planes of order k − 1. Here the blocks are the translates of
R and the points are the elements of G .

In group ring notation a relative difference set satisfies

RR−1 = k + λ(G −N )·

The set D ⊆ G is called partial relative difference set with forbidden set N , if

DD−1 = κ +
∑

g∈G−N

vgg

holds for some 1 ≤ κ ≤ k and 0 ≤ vg ≤ λ for all g ∈ G −N . If D is a relative difference set then ,obviously,
D is also a partial relative difference set.

Two relative difference sets D ,D ′ ⊆ G are called strongly equivalent if they have the same forbidden set
N ⊆ G and if there is g ∈ G and an automorphism α of G such that D ′g−1 = Dα. The same term is applied
to partial relative difference sets.

Let D ⊆ G be a difference set, then the incidence structure with points G and blocks {Dg | g ∈ G} is called
the development of D . In short: devD . Obviously, G acts on devD by multiplication from the right.

If D is a difference set, then D−1 is also a difference set. And devD−1 is the dual of devD . So a group
admitting an operation some structure defined by a difference set does also admit an operation on the
dual structure. We may therefore change the notion of equivalence and take φ to be an automorphism
or an anti-automorphism. Forbidden sets are closed under inversion, so this gives a “weak” sort of strong
equivalence.

4.2 How partial difference sets are represented

Let G be a group. We define an enumeration {g1, . . . , gn} = G and represent D ⊆ G as a list of integers
(where, of course, i represents gi for all 1 ≤ i ≤ n). So the automorphism group of G is represented as
a permutation group of degree n. One of the operations performed most often by methods in RDS is the
calculation of quotients in G . So we calculate a look-up table for this.

This pre-calculation is done by the operation RDS:PermutationRepForDiffsetCalculations. So before you
start generating difference set, call this function and work with the data structure returned by it.

For an exhaustive search, the ordering of G is very important. To avoid generating duplicate partial difference
sets, we would like to represent partial difference sets by sets, i.e. ordered lists. But in fact, RDS does not
assume that partial difference sets are sets. The operations RDS:ExtendedStartSets and RDS:AllDiffsets
assume that the last element of partial difference set is its maximum. But they don’t test it. So if you start
from scratch, these methods generate difference sets which are really sets. Whereas the NoSort versions
disregard the ordering of G and will produce duplicates.

The reason for this seemingly strange behaviour is the following: Assume that we have a normal subgroup
U ≤ G and know that every difference set D ⊆ G contains exactly ni elements from the i th coset modulo
U . Then it is natural to generate difference sets by first searching all partial difference sets of length n1

containing entirely of elements of the first coset modulo U and then proceed with the other cosets.

This method of difference set generation is normally not compatible with the ordering of G . This is why
partial difference sets are not required to be sets. See chapter RDS:An Example Program for an example.

4.3 Basic functions for startset generation

Defining an enumeration of the a group G , every relative difference set may be represented by a list of
integers. Indexing G in this way has the advantage of the automorphism group of G being a permutation
group acting on the index set for G . As relative difference sets are normally calculated in small groups, it is
possible to store a complete multiplication table of the group in terms of the enumeration.

If not stated otherwise, partial difference sets are always considered to be lists of integers. Note that it is
not required for a partial difference set to be a set.

1 I PermutationRepForDiffsetCalculations( group ) O
I PermutationRepForDiffsetCalculations( group, autgrp ) O

For a group group, PermutationRepForDiffsetCalculations(group) returns a record containing:

1. the group .G=group.

2. the sorted list .Glist=Set(group),

3. the automorphism group .A of group,

4. the group .Aac, which is the permutation action of A on the indices of .Glist ,

5. .Ahom=ActionHomomorphism(.A,.Glist),

6. the group .Ai of anti-automorphisms of .group acting on the indices of Glist ,

7. the multiplication table .diffTable of .group in a special form.

.diffTable is a matrix of integers defined such that .difftable[i][j] is the position of Glist[i](Glist[j])^-1)
in Glist with Glist[1]=One(group).

PermutationRepForDiffsetCalculations runs into an error if Set(group)[1] is not equal to One(group).

If autgrp is given, PermutationRepForDiffsetCalculations will not calculate the automorphism group of
group but will take autgrp instead without any test.

If Set(group)[1] is not equal to One(group), then RDS:PermutationRepForDiffsetCalculations returns
an error message. In this case, calculating a permutation representation helps:

gap> G:=SL(2,3);
SL(2,3)
gap> Gdata:=PermutationRepForDiffsetCalculations(G);
Error, smallest element of group is not the identity. Try ‘IsomorphismPermGrou\
p’ called from
<function>( <arguments> ) called from read-eval-loop
Entering break read-eval-print loop ...
you can ’quit;’ to quit to outer loop, or
you can ’return;’ to continue
brk> quit;
gap> G:=Image(IsomorphismPermGroup(G));
Group([ (2,3,5)(6,7,8), (1,2,4,7)(3,6,8,5) ])
gap> Gdata:=PermutationRepForDiffsetCalculations(G);

2 I IsDiffset( diffset, [forbidden], Gdata, [lambda] ) O
I IsDiffset( diffset, [forbidden], group, [lambda] ) O

This function tests if diffset is a relative difference set with forbidden set forbidden and parameter lambda
in the group group. If Gdata is the record calculated by RDS:PermutationRepForDiffsetCalculations,
diffset and forbidden have to be lists of integers. If a group group is given, diffset and forbidden must consist
of elements of this group.

If forbidden is not given, it is assumed to be trivial. If lambda is not given, it is set to 1. Note that 1
(One(group), repectively) must not be element of diffset .

gap> a:=(1,2,3,4,5,6,7);
(1,2,3,4,5,6,7)
gap> IsDiffset([a,a^3],Group(a));
true
gap> IsDiffset([a,a^3],Group(a),2);
false
gap> IsDiffset([a,a^2,a^4],Group(a),2);
true
gap> Gdata:=PermutationRepForDiffsetCalculations(Group(a));;
gap> IsDiffset([2,4],Gdata);
true

3 I IsPartialDiffset( diffset, [forbidden], Gdata, [lambda] ) O
I IsPartialDiffset( diffset, [forbidden], group, [lambda] ) O

This function tests if diffset is a partial relative difference set with forbidden set forbidden and parameter
lambda in the group group. If Gdata is the record calculated by RDS:PermutationRepForDiffsetCalculations,
diffset and forbidden have to be lists of integers. If a group group is given, diffset and forbidden must consist
of elements of this group.

If forbidden is not given, it is assumed to be trivial. If lambda is not given, it is set to 1. Note that 1
(One(group), repectively) must not be element of diffset .

gap> a:=(1,2,3,4,5,6,7);
(1,2,3,4,5,6,7)
gap> IsPartialDiffset([a],Group(a));
true
gap> IsPartialDiffset([a,a^4],Group(a));
false
gap> IsPartialDiffset([a,a^4],Group(a),2);
true

A partial difference set may be converted from a list of group elements to a list of integers using

4 I GroupList2PermList( list, Gdata ) O

converts a list of group elements to integers according to the enumeration given in Gdata.Glist. Here Gdata
is a record containing .diffTable as returned by RDS:PermutationRepForDiffsetCalculations.

The inverse operation is performed by

5 I PermList2GroupList( list, Gdata ) O

converts a list of integers into group elements according to the enumeration given in Gdata.Glist. Here Gdata
is a record containing .diffTable as returned by RDS:PermutationRepForDiffsetCalculations.

gap> G:=DihedralGroup(6);
<pc group of size 6 with 2 generators>
gap> N:=NormalSubgroups(G)[2];
Group([ f2 ])
gap> dat:=PermutationRepForDiffsetCalculations(G);
rec( G := <pc group of size 6 with 2 generators>,
Glist := [ <identity> of ..., f1, f2, f1*f2, f2^2, f1*f2^2 ],
A := <group of size 6 with 2 generators>,
Aac := Group([ (3,5)(4,6), (2,4,6) ]),
Ahom := <action homomorphism>,
Ai := Group([ (3,5), (3,5)(4,6), (2,4,6) ]),
diffTable := [ [ 1, 2, 5, 4, 3, 6 ], [ 2, 1, 6, 3, 4, 5 ],

[ 3, 6, 1, 2, 5, 4 ], [ 4, 5, 2, 1, 6, 3 ],
[ 5, 4, 3, 6, 1, 2 ], [ 6, 3, 4, 5, 2, 1 ] ] )

gap> Nperm:=GroupList2PermList(Set(N),dat);
[ 1, 3, 5 ]

In the following functions the record Gdata has to contain a matrix .diffTable as returned by RDS:PermutationRepForDiffsetCalculations.

6 I NewPresentables( list, newel, table ) O
I NewPresentables( list, newel, Gdata ) O
I NewPresentables( list, newlist, Gdata ) O
I NewPresentables( list, newlist, table ) O

NewPresentables( list,newel,Gdata ) takes a record Gdata as returned by PermutationRepForDiffset-
Calculations(group). For NewPresentables( list,newel,table ), table has to be the multiplication table
in the form of NewPresentables( list,newel,Gdata.diffTable)

The method returns the unordered list of quotients d1newel−1 with d1 ∈ list ∪ {1} (in permutation repre-
sentation).

When used with a list newlist , a list of quotients d1d−1
2 with d1 ∈ list ∪ {1} and d2 ∈ newlist is returned.

7 I AllPresentables( list, table ) O
I AllPresentables( list, Gdata ) O

Let list be a list of integers representing elements of a group defined by Gdata (or table). AllPresenta-
bles( list,table) returns an unordered list of quotients ab−1 for all group elements a, b represented by
integers in list . If 1 ∈ list , an error is issued. The multiplication table table has to be of the form
as returned by RDS:PermutationRepForDiffsetCalculations. And Gdata is a record as calculated by
RDS:PermutationRepForDiffsetCalculations.

gap> G:=CyclicGroup(7);;dat:=PermutationRepForDiffsetCalculations(G);;
gap> AllPresentables([2,3],dat);
[ 2, 3, 7, 2, 7, 6 ]
gap> NewPresentables([2,3],4,dat);
[ 4, 5, 6, 3, 7, 2 ]
gap> AllPresentables([1,2,3],dat);
Error...

8 I RemainingCompletions( diffset, completions[, forbidden], Gdata[, lambda] ) O
I RemainingCompletionsNoSort( diffset, completions[, forbidden], table[, lambda] ) O

For a partial difference set diffset , RemainingCompletions(diffset,completions,Gdata) returns a subset of
the set completions, such that each of its elements may be added to diffset without it loosing the property
to be a partial difference set. Only elements greater than the last element of diffset are returned.

For partial relative difference sets, forbidden is the forbidden set.

RemainingCompletionsNoSort does also return elements from completions which are smaller than diff-
set[Size(diffset)].

gap> G:=CyclicGroup(7);
<pc group of size 7 with 1 generators>
gap> dat:=PermutationRepForDiffsetCalculations(G);;
gap> RemainingCompletionsNoSort([4],[1..7],dat);
[ 2, 3 ]
gap> RemainingCompletionsNoSort([4],[1..7],dat,2);
[ 2, 3, 6, 7 ]
gap> RemainingCompletions([4],[1..7],dat);
[ ]
gap> RemainingCompletions([4],[1..7],dat,2);
[ 6, 7 ]

9 I ExtendedStartsets( startsets, completions, [forbiddenset][, aim], Gdata[, lambda] ) O
I ExtendedStartsetsNoSort( startsets, completions, [forbiddenset][, aim], Gdata[, lambda] ) O

For a set of partial (relative) difference sets startsets, the set of all extensions by one element from completions
is returned. Here an “extension” of a partial diffence set S is a list which has one element more than S and
contains S .

Here completions is a set of elements wich may be appended to the lists in startsets to generate new partial
difference sets. For relative difference sets, the forbidden set forbiddenset must be given. And the integer
aim gives the desired total length, i.e. the number of elements of completions that have to be added to each
startset plus its length. Note that the elements of startset are always extended by one element (if they can
be extended). aim does only tell how many elements from completions you want to add. A partial difference
set is only be extended, if there are enough admissible elements in completions, so if for some S ∈ startsets,
we have less than aim − Size(S ) elements in completions which can be added to S , no extension of S is
returned.

If lambda is not passed as a parameter, it is assumed to be 1.
Note that ExtendedStartsets does use RDS:RemainingCompletions while ExtendedStartsetsNoSort uses
RDS:RemainingCompletionsNoSort. Note that the partial difference sets generated with ExtendedStart-
setsNoSort are not sets (i.e. not sorted). This may result in doing work twice. But it can also be useful,
especially when generating difference sets “coset by coset”.

gap> G:=CyclicGroup(7);;dat:=PermutationRepForDiffsetCalculations(G);;
gap> startsets:=[[2],[4],[6]];;
gap> ExtendedStartsets(startsets,[1..7],dat);
[ [ 2, 4 ], [ 2, 6 ] ]
gap> ExtendedStartsets(startsets,[1..7],3,dat);
[ [ 2, 4 ] ]
gap> ExtendedStartsets(startsets,[1..7],dat,2);
[ [ 2, 3 ], [ 2, 4 ], [ 2, 5 ], [ 2, 6 ], [ 4, 6 ], [ 4, 7 ], [ 6, 7 ] ]
gap> ExtendedStartsetsNoSort(startsets,[1..7],dat);
[ [ 2, 4 ], [ 2, 6 ], [ 4, 2 ], [ 4, 3 ], [ 6, 2 ], [ 6, 5 ] ]

4.4 Brute force methods

The following methods can be used to find (partial) difference sets by brute force. More examples are
contained in chapter RDS:AllDiffsets and OneDiffset

1 I AllDiffsets( [partial], group, [lambda] ) O
I AllDiffsets( partial, [aim], forbidden, group, [lambda] ) O
I AllDiffsets( [partial], Gdata, [lambda] ) O
I AllDiffsets( partial, [aim], forbidden, Gdata, [lambda] ) O
I AllDiffsets( partial, completions, aim, forbidden, Gdata, lambda ) O

Let partial be a list of elements of the group group which form a partial relative difference set with parameter
lambda and forbidden set forbidden (which is also a set of group elements). That means that the every non-
trivial element in the list of quotients in elements of partial occurs at most lambda times and no element
of forbidden is in this set. Then AllDiffsets returns the list of all partial relative difference sets of length
aim with parameter lambda and forbidden set forbidden which contain partial . Only those partial relative
difference sets will be constructed, which start with partial and continue with elements larger than the last
element in partial .
To calculate all difference sets which contain partial as a subset, you can use RDS:AllDiffsetsNoSort.
Note that a difference set is also assumed to contain the identity element, but this does not occur in the
returned lists. So a returned difference set contains aim elements but actually represents a set of length
aim+1, as it still is a partial relative difference set when the identity element is added. If partial is not given
or the empty set, all difference set in the group group are calculated. If lambda is not given, it is set to 1.
Without forbidden, ordinary difference sets are calculated. If aim is not given, it is set to the size of a full
relative difference set with forbidden set forbidden and parameter lambda.
Instead of using a group group, you can also use the data record Gdata returned by RDS:PermutationRepForDiffsetCalculations.
In this case, partial and forbidden must be lists of integers. In the last form, completions must be a list of
integers and AllDiffsets does only extend partial by elements from completions.

2 I AllDiffsetsNoSort( partial, group ) O
I AllDiffsetsNoSort( partial, Gdata ) O
I AllDiffsetsNoSort( partial, [completions], aim, [forbidden], group, [lambda] ) O
I AllDiffsetsNoSort( partial, [completions], aim, [forbidden], Gdata, [lambda] ) O

This calculates all partial relative difference sets which contain the partial relative difference set partial . The
returned value is a set of lists. Each of the returned lists starts with the list partial . If partial is not a partial
relative difference set, the empty list is returned.

Note that despite the name, AllDiffsetsNoSort does not calculate all difference sets as unordered lists. It
just calculates all difference sets which contain partial as a subset.

As it does not only append larger elements to partial , AllDiffsetsNoSort works for all groups.

If called with group rather than Gdata, RDS:AllDiffsets and RDS:AllDiffsetsNoSort call RDS:PermutationRepForDiffsetCalculations.
They then work with sets of integers as difference sets and convert the result back into group notation.

As RDS:PermutationRepForDiffsetCalculations refuses to work if the smallest element of the group is
not 1, this does not always work. So a permutation representation for group is calculated in this case.
However, this is only done for the NoSort version and if partial is empty. Here is an example:

gap> m:=[
> [0,1,0,0,0,0,0],
> [0,0,1,0,0,0,0],
> [0,0,0,1,0,0,0],
> [0,0,0,0,1,0,0],
> [0,0,0,0,0,1,0],
> [0,0,0,0,0,0,1],
> [1,0,0,0,0,0,0]];;
gap> G:=Group(m);
<matrix group with 1 generators>
gap> Order(G);
7
gap> Size(AllDiffsets(G));
6
gap> AllDiffsets([m],G);
Error, smallest element of group is not the identity.
[...]
gap> Size(AllDiffsetsNoSort([m],G));
2

The reason for this is the fact that RDS:AllDiffsets generates difference sets from partial by appending
only elements which are larger than the last element of partial . In a permutation representation, the ordering
will be different from the original one, so GAP refuses to calculate the permutation representation and issues
an error.

RDS:AllDiffsetsNoSort first appends one element regardless of ordering and then only larger ones.

3 I OneDiffset( [partial], group, [lambda] ) O
I OneDiffset( partial, [aim], forbidden, group, [lambda] ) O
I OneDiffset( [partial], Gdata, [lambda] ) O
I OneDiffset( partial, [aim], forbidden, Gdata, [lambda] ) O
I OneDiffset( partial, completions, aim, forbidden, Gdata, lambda ) O

This function works exactly like RDS:AllDiffsets, but stops once a (partial) relative difference set is found.
This (partial) relative difference set is then returned. If no set with the requested property exists, the empty
list is returned.

If OneDiffset is called using Gdata and lists of integers as partial and forbidden, then the returned difference
set is the lexicographically smallest one starting with partial . If the group-form is used and partial is not
empty, OneDiffset does only work, if the smallest element of group is the identity. This is not the case for
matrix groups in general.

4 I OneDiffsetNoSort( partial, group ) O
I OneDiffsetNoSort( partial, Gdata ) O
I OneDiffsetNoSort( partial, [completions], aim, [forbidden], group, [lambda] ) O
I OneDiffsetNoSort( partial, [completions], aim, [forbidden], Gdata, [lambda] ) O

This works exactly as RDS:AllDiffsetsNoSort does, but stops once a set with the desired properties is
found and returns it. If no difference set exists, the empty list is returned.



5
Invariants for

Difference Sets

This chapter contains an important tool for the generation of difference sets. It is called the “coset signature”
and is an invariant for equivalence of partial relative difference sets. For large λ, there is an invariant
calculated by RDS:MultiplicityInvariantLargeLambda. This invariant can be used complementary to the
coset signature and is explained in section RDS:An invariant for large lambda.

Most of the methods explained here are not commonly used. If you do not want to know how coset
signatures work in detail, you can safely skip a large part of this and go straight to the explanation of
RDS:SignatureDataForNormalSubgroups and RDS:ReducedStartsets.

The functions RDS:RDSFactorGroupData, RDS:MatchingFGData will be interesting for you, if you look for
difference sets with the same parameters in different gorups. RDS:SignatureDataForNormalSubgroups and
RDS:SigInvariant

The last section (RDS:Blackbox functions) of this chapter has some functions which allow the user to
use coset signatures with even less effort. But be aware that these functions make choices for you that you
probably do not want if you do very involved calculations. In particular, the coset signatures are not stored
globally and hence cannot be reused. For a demonstration of these easy-to-use functions, see chapter RDS:A
basic example

5.1 The Coset Signature

Let R ⊆ G be a (partial) relative difference set (for definition see RDS:Introduction) with forbidden set
N ⊆ G . Let U ≤ G be a normal subgroup and C = {g1, . . . , g|G:U |} be a system of representatives of G/U .

The intersection number of R with Ugi is defined as vi = |R ∩Ugi |. For every normal subgroup U ≤ G the
multiset {|R ∩Ugi |: gi ∈ C} is called “coset signature of R (relative to U )”.

Let D ⊆ G be a relative difference set and {v1, . . . , v|G:U |} its coset signature. Then the following equations
hold (see [Bru55],[Röd06]): ∑

vi = k∑
v2
i = λ(|U | − |U ∩N |) + k∑

j vj vij = λ(|U | − |giU ∩N |) for gi 6∈ U

where vij = |D ∩ gigjU |. If the forbidden set N is a subgroup of G we have |giU ∩N | is either 0 or equal to
|U ∩N |.
Given a group G , the forbidden set N ⊆ G and some normal subgroup U ≤ G , the right sides of this
equations are known. So we may ask for tuples (v1, . . . , v|G:U |) solving this system of equations. Of course,
we index the vi with the elements of G/U , so the last equation poses conditions to the ordering of the tuple
(v1, . . . , v|G:U |).

So we call any multiset {v1, . . . , v|G:U |} solving the above equations an “admissible signature” for U .

1 I CosetSignatureOfSet( set, cosets ) F

CosetSignatureOfSet( set,cosets) returns the ordered list of intersection numbers of set . That is, the
size of the intersection of set with each Element of cosets.

Note that it is not tested, if cosets is really a list of cosets. CosetSignatureOfSet( set,cosets) works for
any List set and any list of lists cosets. So be careful!

gap> G:=SymmetricGroup(5);;
gap> A:=AlternatingGroup(5);;
gap> CosetSignatureOfSet([(1,2),(1,5),(1,2,3)],RightCosets(G,A));
[ 1, 2 ]
gap> CosetSignatureOfSet([(1,2),(1,5),(1,2,3)],[A]);
[ 1 ]
gap> CosetSignatureOfSet([(1,2),(1,5),(1,2,3)],[[(1,2),(1,2,3)],[(3,2,1)]]);
[ 0, 2 ]

2 I CosetSignatures( Gsize, Usize, diffsetorder ) O
I CosetSignatures( Gsize, Nsize, Usize, Intersectsizes, k, lambda ) O

CosetSignatures( Gsize,Usize,diffsetorder) returns all Gsize/Usize tuples such that the sum of the
squares of each tuple equals Usize+diffsetorder . And the sum of each tuple equals diffsetorder+1.

These are necessary conditions for signatures of difference sets and normal subgroups of order Usize in
groups of order Gsize (see RDS:The Coset Signature).

CosetSignatures( Gsize,Nsize,Usize,Intersectsizes,k,lambda) Calculates all multiset meeting some con-
ditions for signatures of relative difference sets and normal subgroups of order Usize in groups of order Gsize
(see RDS:The Coset Signature). Here Nsize is the size of the forbidden group, Intersectsizes is a list of
integers determining the size of the intersection of the forbidden set and the normal Subgroup of order Usize.
The pararmeters k and lambda are the usual ones for designs. CosetSignatures returns a list containing
one pair for each entry i of Intersectsizes. The first entry of this pair is [Gsize,Nsize,Usize, i , k , lambda] and
the second one is a list of admissible signatures with these parameters.

gap> CosetSignatures(256,16,64,[1,4,8,16],17,1);
[ [ [ 256, 16, 64, 1, 17, 1 ], [ ] ],
[ [ 256, 16, 64, 4, 17, 1 ], [ [ 3, 4, 4, 6 ] ] ],
[ [ 256, 16, 64, 8, 17, 1 ], [ [ 4, 4, 4, 5 ] ] ],
[ [ 256, 16, 64, 16, 17, 1 ], [ ] ] ]

#And for an ordinary difference set of order 16.
gap> CosetSignatures(273,1,39,[1],17,1);
[ [ [ 273, 1, 39, 1, 17, 1 ],
[ [ 0, 1, 2, 3, 3, 4, 4 ], [ 0, 2, 2, 2, 3, 3, 5 ],
[ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ],
[ 1, 1, 2, 2, 2, 4, 5 ] ] ] ]

3 I TestSignatureLargeIndex( sig, group, Normalsg[, factorgrp] ) O

this does only work for ordinary difference sets, not for relative difference sets in general

TestSignatureLargeIndex(sig,group,Normalsg[,factorgrp]) tests if sig meets some necessary conditions
of RDS:The Coset Signature to be a signature for a difference set in group for the normal subgroup
Normalsg . factorgrp is the factorgroup group/Normalsg . The returned value is true or false resp.

4 I TestSignatureCyclicFactorGroup( sig, Nsize ) O

This does only work for ordinary difference sets, not for relative difference sets in general

TestSignatureCyclicFactorGroup(sig,Nsize) test if sig meets meets some necessary conditions of RDS:The
Coset Signature to be a signature for a difference set in some group, which has a normal subgroup of size
Nsize such that the factor group is cyclic. The returned value is true or false resp.

5 I TestedSignatures( sigs, group, Normalsg[, maxtest][, moretest] ) O

this does only work for ordinary difference sets, not for relative difference sets in general

Let sigs be a list of possible signatures as returned by RDS:CosetSignatures. Let Normalsg be a subgroup
of group. For each signature in sigs, the necessary conditions described in RDS:The Coset Signature are
tested to decide if the signature can be a signature of a difference set in group for for the normal subgroup
Normalsg .

As this involves computation for all permutations of the signature, this can be very costly. The argument
maxtest determines how many permutations are admissible. If maxtest=0, all signatures are tested, regardless
of how much work is necessary for this. If a signature has too many permutations, it is returned without test.
Even though it is not wise, maxtest=0 is the default option. If InfoLevel(InfoRDS) is at least 2, information
about skipped signatures is echoed.

If the boolean value moretest is false and all signatures in sigs but the last one are found to be not admissible,
the last one is returned without test. This saves the time to test the last signature, but if chances are that
there is no difference set in group, this may also give away a chance to find out early (every difference set
has signatures, so no admissible signature means that no difference set can exist). Default is true.

TestedSignatures calls RDS:TestSignatureCyclicFactorGroup or RDS:TestSignatureLargeIndex and
returns a sublist of sigs.

gap> G:=SmallGroup(273,2);;
gap> N:=First(NormalSubgroups(G),g->Order(g)=39);
Group([ f1, f3 ])
gap> sigs:=CosetSignatures(273,1,39,[1],17,1);
[ [ [ 273, 1, 39, 1, 17, 1 ],
[ [ 0, 1, 2, 3, 3, 4, 4 ], [ 0, 2, 2, 2, 3, 3, 5 ],
[ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ],
[ 1, 1, 2, 2, 2, 4, 5 ] ] ] ]

gap> TestedSignatures(sigs[1][2],G,N);
[ [ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ] ]

6 I TestedSignaturesRelative( sigs, fgdata, [, maxtest][, moretest] ) O

TestedSignaturesRelative takes a list sigs of lists of integers and returns a those which may be signatures
of relative difference sets with forbidden set.

fgdata is a record as returned by RDSFactorGroupData(U ,N ,lambda,Gdata) If maxtest is set, a signature
s is only tested if NrPermutationsList(s) is less than maxtest if maxtest is set to 0, all signatures are
tested this is the default. If moretest is tue, a signature is tested even if it is the only one left. This means
we do not assume that there must be an admissable signature at all. The default for moretest is true.

7 I SigInvariant( diffset , data ) O

Given a partial relative difference set diffset and a list of records with entries cosets and sigs. Here cosets is
a full list of cosets and sigs is a list of signatures that may occur for relative difference sets.

For each record rec in data, the intersection numbers of diffset with the cosets of rec.cosets are computed
stored in a set sig . If none of the signatures in rec.sigs is pointwise greater or equal sig , SigInvariant(
diffset,data) returns ‘fail. Otherwise sig is added to a list of signatures that is returned.

Note the returned invariant is that of diffset ∪ {1}. The output from SignatureDataForNormalSubgroups
can be used as data.

gap> G:=SmallGroup(273,2);
<pc group of size 273 with 3 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> N:=First(NormalSubgroups(G),g->Order(g)=39);
Group([ f1, f3 ])
gap> sigs:=CosetSignatures(273,1,39,[1],17,1);
[ [ [ 273, 1, 39, 1, 17, 1 ],
[ [ 0, 1, 2, 3, 3, 4, 4 ], [ 0, 2, 2, 2, 3, 3, 5 ],
[ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ],
[ 1, 1, 2, 2, 2, 4, 5 ] ] ] ]

gap> TestedSignatures(sigs[1][2],G,N);
[ [ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ] ]
gap> sigs:=TestedSignatures(sigs[1][2],G,N);
[ [ 1, 1, 1, 2, 4, 4, 4 ], [ 1, 1, 1, 3, 3, 3, 5 ] ]
gap> ## calculate cosets in permutation notation:
gap> rc:=List(RightCosets(G,N),i->GroupList2PermList(Set(i),Gdata));;
gap> data:=[rec(cosets:=rc,sigs:=sigs)];;
gap> SigInvariant([3,4,5],data);
[ [ [ 0, 0, 0, 0, 0, 1, 3 ], 1 ] ]

For an example using RDS:SignatureDataForNormalSubgroups see the example after RDS:ReducedStartsets
below.

8 I SignatureDataForNormalSubgroups( Normals, globalSigData, forbiddenSet, Gdata, parameters ) O

Let Gdata be a record as returned by RDS:PermutationRepForDiffsetCalculations. Let Normals be a
list of normal subgroups of Gdata.G , and forbiddenSet the forbidden set (as set of group elements or group).

parameters must be a list of length 4 of the form [k,lambda,maxtest,moretest] with k the length of the relative
difference set to be constructed and lambda the parameter as always. maxtest and moretest are passed to
TestedSignaturesRelative and must be set.

SignatureDataForNormalSubgroups returns a list containing one record for each group U in Normals. This
record contains:

1. the subgroup U named .subgroup

2. the signatures .sigs for U

3. the cosets .cosets modulo U as lists of integers

Moreover, the list globalSigData is used to store global information which can be reused with other groups.
The i th entry of globalSigData is a list of records that contains all known information about subgroups of
order i . Each of these records has the following components:

1. .cspara the parameters for RDS:CosetSignatures

2. .sigs the output of RDS:CosetSignatures when the input is .cspara

3. .fgsigs a list of records containing data about factor groups with parameters .cspara:

3.1. .fg the factor group

3.2. .fgaut the automorphism group of .fg

3.3. .Nfg the image of the forbidden set N under the natural epimorphism to .fg

3.4. .fgintersect the pairs [g , |g ∩N |] for all g in .fg . Here N is the forbidden set.

3.5. .sigs the known admissible signatures (this is a subset of the set in number 2. of course)

The list globalSigData can be used if different groups are studied. If a group has a normal subgroup with
parameters (in the sense of .cspara) listed in globalSigData, the signatures from a previous calculation may
be used. Of course, the factor groups have to be checked first. This check is done with RDS:MatchingFGData
or RDS:MatchingFGDataNonGrp.

So the second run of SignatureDataForNormalSubgroups with the same parameters and different Gdata
and Normals will normally be much faster, as the signatures are already stored in globalSigData. Note that
maxtest and moretest are not stored. So a second run with larger maxtest will not result in a recalculation
of signatures.

gap> G:=CyclicGroup(57);
<pc group of size 57 with 2 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> SignatureDataForNormalSubgroups(NormalSubgroups(Gdata.G),sigdata,
> [One(Gdata.G)],Gdata,[8,1,10^6,true]); # for ordinary diffset of order 7.
[ rec( subgroup := Group([ f1*f2^6 ]),

sigs := [ [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 1, 1, 1, 1, 1, 2 ] ],
cosets := [ [ 1, 20, 40 ], [ 3, 23, 43 ], [ 6, 26, 46 ], [ 9, 29, 49 ],

[ 12, 32, 52 ], [ 15, 35, 55 ], [ 18, 38, 57 ],
[ 4, 21, 41 ], [ 7, 24, 44 ], [ 10, 27, 47 ],

[ 13, 30, 50 ], [ 16, 33, 53 ], [ 19, 36, 56 ],
[ 2, 22, 39 ], [ 5, 25, 42 ], [ 8, 28, 45 ], [ 11, 31, 48 ],
[ 14, 34, 51 ], [ 17, 37, 54 ] ] ),
rec( subgroup := Group([ f2 ]), sigs := [ [ 1, 3, 4 ] ],

cosets := [ [ 1, 3, 6, 9, 12, 15, 18, 21, 24, 27, 30, 33, 36, 39, 42,
45, 48, 51, 54 ],

[ 2, 5, 8, 11, 14, 17, 20, 23, 26, 29, 32, 35, 38, 41, 44, 47, 50,
53, 56 ],

[ 4, 7, 10, 13, 16, 19, 22, 25, 28, 31, 34, 37, 40, 43, 46, 49,
52, 55, 57 ] ] ) ]

gap> Filtered([1..Size(sigdata)],i->IsBound(sigdata[i]));
[ 3, 19 ]
gap> Size(sigdata[3]);
2
gap> sigdata[3][1].cspara;sigdata[3][2].cspara;
[ 57, 1, 3, 1, 7, 1 ]
[ 57, 1, 3, 1, 8, 1 ]

The following three functions are used by RDS:SignatureDataForNormalSubgroups. If you do not want to
write your own function for signature management, you might not need them.

9 I RDSFactorGroupData( U , N , lambda, Gdata ) O

takes the subgroup U of G , the forbidden set N as a subgroup or subset of G and the record of data Gdata
as returned by PermutationRepForDiffsetCalculations(G) and returns a record containing

.fg the factor group modulo U

.fglist the factor group as a strictly ordered list

.cosets the cosets modulo U as lists of integers

.lambda the parameter lambda as passed to the function

.Usize the size of U

.fgaut the automorphism group of .fg

.Nfg the image of N in .fg

.fgintersect a list of pairs such that the i th entry is the pair consisting of .fg[i] and the size of the intersection of .fg
with .Nfg as cosets modulo U .

.intersectshort ist just the second component of .fgintersect .

10 I MatchingFGDataNonGrp( fgdatalist, fgmatchdata ) O

Let fgdatalist be a list of records and fgmatchdata a record with components .fg , .Nfg and .fgintersect as
returned by RDS:RDSFactorGroupData. Then MatchingFGDataNonGrp returns the entry of fgdatalist that
defines the same admissible signatures as fgmatchdata. If no such entry exists, fail is returned.

The forbidden set N is not assumed to be a group.

11 I MatchingFGData( fgdatalist, fgmatchdata ) O

Let fgdatalist be a list of records and fgmatchdata a record with components .fg , .Nfg , .fgintersect and .fgaut
as returned by RDS:RDSFactorGroupData. Then MatchingFGDataNonGrp returns the entry of fgdatalist that
defines the same admissible signatures as fgmatchdata. If no such entry exists, fail is returned.

Here the forbidden set N has to be a group.

12 I ReducedStartsets( startsets, autlist, csdata, Gdata ) O
I ReducedStartsets( startsets, autlist, func, Gdata ) O

Let startsets be a set of partial relative difference sets, autlist a list of permutation groups and Gdata
record returned by PermutationRepForDiffsetCalculations. Then ReducedStartsets partitions the list
startsets according to the values of the function func and performs a test for equivalence on the elements of
the partition. The list returned is a sublist of startsets of pairwise non-equivalent partial relative difference
sets if func is an invariant for partial relative difference sets. All elements for which func returns fail are
discarded.

If a list csdata of records as used for RDS:SigInvariant (i.e. containing .cosets and .signatures) is pased,
then ReducedStartsets uses RDS:SigInvariant for func.

gap> G:=CyclicGroup(57);
<pc group of size 57 with 2 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> cosetsigs:=SignatureDataForNormalSubgroups(NormalSubgroups(Gdata.G),
> sigdata, [One(Gdata.G)],Gdata,[8,1,10^6,true]);;
gap> SigInvariant([3,4,5,9],cosetsigs);
[ [ [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 1, 1, 1, 1 ], 1 ], [ [ 1, 1, 3 ], 1 ] ]
gap> ssets:=AllDiffsets([],2,[],Gdata);;
gap> Size(ssets);
1458
gap> Size(ReducedStartsets(ssets,[Group(())],cosetsigs,Gdata));
#I Size 1458
#I 5/ 0 @ 0:00:00.126
486
gap> Size(ReducedStartsets(ssets,[Gdata.Ai],cosetsigs,Gdata));
#I Size 1458
#I 5/ 0 @ 0:00:00.123
17

13 I MaxAutsizeForOrbitCalculation V

In RDS:ReducedStartsets, a bound is needed to decide if Orbit or RepresentativeAction should be
used. If the group is larger than MaxAutsizeForOrbitCalculation@RDS , RepresentativeAction is used.
The default value for maxAutsizeForOrbitCalculation is 5 ∗ 106. If you want to change it, you will have
to edit the file sigs.gd.

5.2 An invariant for large lambda

1 I MultiplicityInvariantLargeLambda( set, Gdata ) O

Let set be a partial relative difference set with λ > 1. Set P:=AllPresentables(set,Gdata) then the set
of multiplicities of P is an invariant for partial relative difference sets.

MultiplicityInvariantLargeLambda returns a list in a form as Collected does.

gap> G:=CyclicGroup(7);;Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> AllPresentables([2,3],Gdata);
[ 2, 3, 7, 2, 7, 6 ]
gap> MultiplicityInvariantLargeLambda([2,3],Gdata);
[ [ 1, 2 ], [ 2, 2 ] ]

(Read this output as: two elements occur once and two occur twice).

This invariant can be used for RDS:ReducedStartsets complementary to the signature invariant by defining

gap> partfunc:=function(list)
> local sig;
> if sig=fail
> then return fail;
> fi;
> return [MultiplicityInvariantLargeLambda(list,Gdata),SigInvariant(list,sigdata)];
> end;
function( list ) ... end

partfunc can then be passed to RDS:ReducedStartsets. Of course, sigdata has to be the list of records
defining the coset signatures.

5.3 Blackbox functions

Here are a few functions used in chapter RDS:A basic example. These are meant as black boxes for quick
tests. Some of them make choices for you which might not be suitable to the chase you consider, so for
serious studies, consider using the more complicated-looking functions above (an example for this comprises
chapter RDS:An Example Program).

1 I SignatureData( Gdata, forbiddenSet, k, lambda, maxtest ) F

Let Gdata be a record as returned by RDS:PermutationRepForDiffsetCalculations. Let forbiddenSet the
forbidden set (as set or group).

k is the length of the relative difference set to be constructed and lambda the usual parameter. maxtest is
the Then SignatureData calls RDS:SignatureDataForNormalSubgroups for normal subgroups of order at
least RootInt(Gdata.G). Here maxtest is an integer which determines how many permutations of a possible
signature are checked to be a sorted signature. Choose a value of at least 105. Larger numbers here normaly
result in better results when generating difference sets (making reduction more effective).

SigntureData chooses normal subgroups of Gdata.G and uses RDS:SignatureDataForNormalSubgroups
to calculate signature data. The global data generated by RDS:SignatureDataForNormalSubgroups is just
discarded.

gap> G:=CyclicGroup(57);;Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> sigdat:=SignatureData(Gdata,[One(Gdata.G)],8,1,10^5);
[ rec( subgroup := Group([ f2 ]), sigs := [ [ 1, 3, 4 ] ],

cosets := [ [ 1, 3, 6, 9, 12, 15, 18, 21, 24, 27, 30, 33, 36, 39, 42, 45, 48, 51, 54 ],
[ 2, 5, 8, 11, 14, 17, 20, 23, 26, 29, 32, 35, 38, 41, 44, 47, 50, 53, 56 ],
[ 4, 7, 10, 13, 16, 19, 22, 25, 28, 31, 34, 37, 40, 43, 46, 49, 52, 55, 57 ] ] ) ]

2 I NormalSgsHavingAtMostNSigs( sigdata, n, lengthlist ) F

Let sigdata be a list as returned by RDS:SignatureDataForNormalSubgroups, an integer n and a list of
integers lengthlist . NormalSgsHavingAtMostNSigs filters sigdata and returns a list of records with compo-
nents .subgroup and .sigs is returned, such that for every entry .subgroup is a normal subgroup of index in
lengthlist having at most n signatures.

3 I SuitableAutomorphismsForReduction( Gdata, normalsg ) F

Given a normal subgroup normalsg of Gdata.G , the function returns a list containing the group of auto-
morphisms of Gdata.G which stabilizes all cosets modulo normalsg . This group is returned as a group of
permutations on Gdata.Glist (which is actually the right regular representation). The returned list can be
used with RDS:StartsetsInCoset.

4 I StartsetsInCoset( ssets, coset, forbiddenSet, aim, autlist, sigdat, Gdata, lambda ) F

Assume, we want to generate difference sets “coset by coset” modulo some normal subgroup. Let ssets
be a (possibly empty) set of startsets, coset the coset from which to take the elements to append to the
startsets from ssets. Furthermore, let aim be the size of the generated partial difference sets (that is, the
size of the elements from ssets plus the number of elements to be added from coset). Let autlist be a list
of groups of automorphisms (in permutation representation) to use with the reduction algorithm. Here the
output from SuitableAutomorphismsForReduction can be used. And Gdata and sigdat are the records as
returned by RDS:PermutationRepForDiffsetCalculations and RDS:SignatureDataForNormalSubgroups
(or RDS:SignatureData, alternatively). The parameter lambda is the usual one for difference sets (the
number of ways of expressing elements outside the forbidden set as quotients).

Then StartsetsInCoset returns a list of partial difference sets (a list of lists of integers) of length aim.

The list of permutation groups autlist is used for equivalence testing. Each equivalence test is performed
calculating equivalence with respect to the first group, one element per equivalence class is retained and the
equivalence test is repeated using the second group from autlist ... Using an ascending list of automorphism
groups can speed up the process of equivalence testing.

gap> G:=CyclicGroup(57);;Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> sigdat:=SignatureData(Gdata,[One(Gdata.G)],8,1,10^5);;
gap> N:=First(NormalSubgroups(G),n->Size(n)=19);
gap> auts:=SuitableAutomorphismsForReduction(Gdata,N);
[ <permutation group of size 18 with 3 generators> ]
gap> g:=One(G);;while g in N do
> g:=Random(G);
> od;
gap> coset:=GroupList2PermList(Set(RightCoset(N,g)),Gdata);
[ 2, 5, 8, 11, 14, 17, 20, 23, 26, 29, 32, 35, 38, 41, 44, 47, 50, 53, 56 ]
gap> Size(StartsetsInCoset([],coset,[],4,auts,sigdat,Gdata,1));
#I Size 19
#I 1/ 0 @ 0:00:00.003
#I Size 26
#I 1/ 0 @ 0:00:00.001
#I -->10 @ 0:00:00.004

#I Size 88
#I 1/ 0 @ 0:00:00.003
#I -->45 @ 0:00:00.018
#I Size 125
#I 1/ 0 @ 0:00:00.006
#I -->64 @ 0:00:00.031
64
gap> Size(StartsetsInCoset([],coset,[],4,[Group(())],sigdat,Gdata,1));
#I Size 19
#I 1/ 0 @ 0:00:00.000
#I Size 136
#I 1/ 0 @ 0:00:00.004
#I -->136 @ 0:00:00.024
#I Size 648
#I 1/ 0 @ 0:00:00.021
#I -->648 @ 0:00:00.310
#I Size 1140
#I 1/ 0 @ 0:00:00.036
#I -->1140 @ 0:00:00.980
1140



6 An Example Program

Here is a similar example to that in chapter RDS:A basic example. But now we do a little more handwork
to see how things work. Now we will calculate the relative difference sets of “Dembowski-Piper type d” and
order 16. These difference sets represent projective planes which admit a quasiregular collineation group
such that the fixed structure is an anti-flag. See [DP67], [Dem68] or [Röd06] for details.

To have a little more output, you may want to increase RDS:InfoRDS:

gap> SetInfoLevel(InfoRDS,3);

First, define some parameters and calculate the data needed:

gap> k:=16;;lambda:=1;;groupOrder:=255;; #Diffset parameters
gap> forbiddenGroupOrder:=15;;
gap> maxtest:=10^6;; #Bound for sig testing
gap> G:=CyclicGroup(groupOrder);
<pc group of size 255 with 3 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> MakeImmutable(Gdata);;

Now the forbidden group is calculated in a very ineffective way. Then we calculate admissible signatures. As
there are only few normal subgroups in G , we calculate them all. For other groups, one should choose more
wisely.

gap> N:=First(NormalSubgroups(Gdata.G),i->Size(i)=forbiddenGroupOrder);
Group([ f1*f3^9, f2*f3^10 ])
gap> globalSigData:=[];;
gap> normals:=Filtered(NormalSubgroups(Gdata.G),n->Size(n) in [2..groupOrder-1]);;
gap> sigdat:=SignatureDataForNormalSubgroups(normals,globalSigData,
> N,Gdata,[k,lambda,maxtest,true]);;

The last step gives better results, if a larger maxtest is chosen. But it also takes more time. To find a suitable
maxtest , the output of RDS:SignatureDataForNormalSubgroups can be used, if InfoLevel(InfoRDS) is at
least 2. Note that for recalculating signatures, you will have to reset globalSigData to []. Try experimenting
with maxtest to see the effect of signatures for the generation of startsets.

Now examine the signatures found. Look if there is a normal subgroup which has only one admissible
signature (of course, you can also use RDS:NormalSgsHavingAtMostNSigs here):

gap> Set(Filtered(sigdat,s->Size(s.sigs)=1 and Size(s.sigs[1])<=5),i->i.sigs);
[ [ [ 0, 4, 4, 4, 4 ] ], [ [ 4, 4, 8 ] ] ]

Great! we’ll take the subgroup of index 3. The cosets modulo this group will be used to generate startsets
and we assume that the trivial coset contains 8 elements of the difference set. So we generate startsets in
U and make a first reduction. For this, we need U and N as lists of integers (recall that difference sets are
asumed to be lists of integers). We will call these lists Up and Np. Furthermore, we will have to choose a
suitable group of automorphisms for reduction. As G is cyclic, we may take Gdata ·Aac here. A good choice

is the stabilizer of all cosets modulo U . Yet sometimes larger groups may be possible. For example if we
want to generate start sets in U and then do a brute force search. In this case, we may take the stabilizer
of U for reduction.

gap> U:=First(sigdat,s->s.sigs=[ [ 4, 4, 8 ] ]).subgroup;
Group([ f2, f3 ])
gap> cosets:=RightCosets(G,U);
gap> U1:=cosets[2];;U2:=cosets[3];;
gap> Up:=GroupList2PermList(Set(U),Gdata);;
gap> Np:=GroupList2PermList(Set(N),Gdata);
[ 1, 12, 25, 43, 78, 97, 115, 116, 134, 151, 169, 188, 207, 238, 249 ]
gap> comps:=Difference(Up,Np);;
gap> ssets:=List(comps,i->[i]);;
gap> ssets:=ReducedStartsets(ssets,[Gdata.Aac],sigdat,Gdata.diffTable);
#I Size 80
#I 2/ 0 @ 0:00:00.061
[ [ 3 ], [ 4 ] ]

Observe that 1 is assumed to be element of every difference set and is not recorded explicitly. So the partial
difference sets represented by ssets at this point are [ [ 1, 3 ], [ 1, 4 ] ]. Now the startsets are ex-
tended to size 7 using elements of Up. The runtime varies depending on the output of RDS:SignatureDataForNormalSubgroups
and hence on maxtest .

gap> repeat
> ssets:=ExtendedStartsets(ssets,comps,Np,7,Gdata);
> ssets:=ReducedStartsets(ssets,[Gdata.Aac],sigdat,Gdata.diffTable);;
> until ssets=[] or Size(ssets[1])=7;
#I Size 133
#I 3/ 0 @ 0:00:00.133
#I Size 847
#I 4/ 0 @ 0:00:00.949
#I Size 6309
#I 4/ 0 @ 0:00:07.692
#I Size 21527
#I 5/ 0 @ 0:00:28.337
#I Size 15884
#I 4/ 0 @ 0:00:21.837
#I Size 1216
#I 4/ 0 @ 0:00:01.758
gap> Size(ssets);
192

At a higher level of RDS:InfoRDS, the number of start sets which are discarded because of wrong signatures
are also shown. Now the cosets are changed. Here we use the NoSort version of RDS:ExtendedStartsets.
This leads to a lot of start sets in the first step. If the number of start sets in U is very large, this could be
too much for a reduction. Then the only option is using the brute force method. But also for the brute force
search, RDS:ExtendedStartSetsNoSort must be called first (remember that we chos an enumeration of G
and assume the last element from each startset to be the largeset “interesting” one for further completions).



31

gap> comps:=Difference(GroupList2PermList(Set(U1),Gdata),Np);;
gap> ssets:=ExtendedStartsetsNoSort(ssets,comps,Np,11,Gdata);;
gap> ssets:=ReducedStartsets(ssets,[Gdata.Aac],sigdat,Gdata.diffTable);;
#I Size 8640
#I 9/ 0 @ 0:00:14.159
gap> Size(ssets);
6899

And as above, we continue:

repeat
ssets:=ExtendedStartsets(ssets,comps,Np,11,Gdata);
ssets:=ReducedStartsets(ssets,[Gdata.Aac],sigdat,Gdata.diffTable);;

until ssets=[] or Size(ssets[1])=11;
comps:=Difference(GroupList2PermList(Set(U2),Gdata),Np);
RaiseStartSetLevelNoSort(ssets,comps,Np,15,Gdata);
repeat

ssets:=ExtendedStartsets(ssets,comps,Np,15,Gdata);
ssets:=ReducedStartsets(ssets,[Gdata.Aac],sigdat,Gdata.diffTable);;

until ssets=[] or Size(ssets[1])=15;



7 Ordered Signatures

In this chapter, we will discuss two methods to calculate ordered signatures. The first one can be used for
relative difference sets with forbidden set, while the second one does only work for ordinary difference sets.

The methods introduced here can only be used in some special cases.

7.1 Ordered signatures by quotient images

Let D ⊆ G be a relative difference set with parameters (v/n,n, k , λ) and forbidden set N ⊆ G . Let U ≤ G
be a normal subgroup such that U ⊆ N .

Then the coset signature (v1, . . . , v|G:U |) of D has only the entries 1 (k - times) and 0 (|G : U | − k - times).
And as in chapter RDS:Invariants for Difference Sets we have

∑
j

vj vij = λ(|U | − |giU ∩N |) for gi 6∈ U

where vij = |D ∩ gigjU |. If the forbidden set N is a subgroup of G we have |giU ∩N | is either 0 or equal to
|U ∩N | = |U |.
Let φ:G → G/U be the canonical epimorphism. Then Dφ is a relative difference set in G/U with forbidden
set N φ and parameters (v/n,n/|U |, k , |U |λ).

So the ordered signatures with respect to U are equivalent to the relative difference sets in G/U . Observe that
we may not apply reduction in G/U using the full automorphismgroup of G/U but only the group induced
by the stabiliser of U in the automorphism group of G . This is due to the fact that we use an “induced”
notion of equivalence in G/U because we are interested in signatures and not primarily in difference sets in
G/U .

1 I NormalSgsForQuotientImages( forbidden, Gdata ) O

calculates all normal subgroups of Gdata.G which lie in forbidden. The returned value is a list of normal
subgroups which define pairwise non-isomorphic factor groups.

2 I DataForQuotientImage( normal, forbidden, k, lambda, Gdata ) O

Let Gdata be the usual record for a group G . And let k and lambda be the parameters of the relative
difference set we want to find. Let then forbidden be the forbidden set (as a group or a list of group elements
or integers) and normal a normal subgroup of G which is contained in forbidden.

Then DataForQuotientImage returns a record containing the record .Gdata of the factor group G/U where
the automorphism group is the one induced by the stabiliser of normal in the automorphism group of G .
Furthermore the returned record contains the forbidden set .forbidden in G/U and the new parameter
.lambda for the difference set in G/U .

The data returned by RDS:DataForQuotientImage can be used to calculate difference sets in G/U in the
way outlined in chapter RDS:A basic example. A quotient image of a relative difference set has a larger

λ than the initial difference set. So RDS:MultiplicityInvariantLargeLambda can be used as in invariant
here (see RDS:An invariant for large lambda)

After all difference sets are known, they must be converted into ordered signatures. This is done by the
following function:

3 I OrderedSigsFromQuotientImages( fGroupData, qimages, forbidden, normal, Gdata ) O

Let Gdata be the usual record for a group G and normal a normal subgroup of G which lies in the
forbidden set forbidden. Let then fGroupData be the record .Gdata describing G/normal as returned by
RDS:DataForQuotientImage and qimages a set of difference sets in G/normal .

Then OrderedSigsFromQuotientImages returns a record containing a list of ordered signatures .orderedSigs
and a list of cosets .cosets as well as the factor group .fg defined by fGroupData and its full automorphism
group fgaut and the image of forbidden in .fg is returned as .Nfg .

4 I MatchingFGDataForOrderedSigs( forbidden, Gdata, Normalsgs, fgdata ) O

Let fgdata be a list of records of the form returned by RDS:OrderedSigsFromQuotientImages and Normalsgs
a list of normal subgroups of the group Gdata.G . Furthermore let forbidden be the forbidden set as a list of
group elements or integers or a subgroup of Gdata.G .

Then MatchingFGDataForOrderedSigs retruns all elements of fgdata which match a normal subgroup of
Normalsgs. The returned value is a record containing the normal subgroup .normal from Normalsgs, the
record .sigdata from fgdata and a homomorphism .hom which maps Gdata.G onto .sigdata.Gdata.G and
takes forbidden to .sigdata.Nfg .

5 I OrderedSigInvariant( set, data ) O

does the same as RDS:SigInvariant, but for ordered signatures. Here data has to be a list of records contain-
ing ordered signatures called .orderedSigs and cosets .cosets just as returned by RDS:OrderedSigsFromQuotientImages.

Assume we have calculated ordered signatures and have stored them in a record .osigs and a list normal-
SubgroupsData as returned by RDS:SignatureData containing the admissible signatures. A function for
partitioning partial relative difference sets as required by RDS:ReducedStartsets can be defined as follows:

partitionfunc:=function(list)
local si, osi;
si:=SigInvariant(Union(list,[1]),normalSubgroupsData);
osi:=OrderedSigInvariant(Union(list,[1]),[osigs]);
if osi=fail or si=fail
then
return fail;

else
return si;

fi;
end;

7.2 Ordered signatures using representations

This section contains some methods for ordered signatures in ordinary difference sets. Unfortunately, these
methods are not as comfortable as those for unordered signatures. The reason for this is simply that I didn’t
have any time to tie them together to high-level functions. If you need help here, don’t hesitate to contact
me.

7.3 Definition

Let R ⊆ G be a (partial) ordinary difference set (for definition see RDS:Introduction). Let U ≤ G be a
normal subgroup and C = {g1, . . . , g|G:U |} be a system of representatives of G/U .

As in RDS:The Coset Signature we may define the coset signature of R relative to U .

Let U = g1, . . . , g|G:U | be an enumeration of G/U . An “admissible ordered signature” for U is a tuple
(v1, . . . , v|G:U |) such that ∑

vi = k∑
v2
i = λ(|U | − 1) + k∑

j vj vij = λ(|U | − 1) for gi 6∈ U

holds where we index the vi by elements of G/U , so vi = vgi and write vij = vgi gj . Observe that the third
equation is a restriction on the ordering of the tuple (v1, . . . , v|G:U |). If v is an admissible ordered signature,
then the multiset of v is an unordered signature.

Getting ordered admissible signatures from unordered ones can be done by taking all permutations of
the unordered signature and verifying the above equations. Obviously, this method isn’t very satisfying
(nevertheless, the methods for testing unordered signatures from section RDS:The Coset Signature do this
to find out if there is an ordered signature at all. Except that they stop when they find an ordered signature).

For ordinary difference sets in extensions of semidirect products of cyclic groups, ordered signatures may be
calculated a lot easier (see [Röd06] for details).

7.4 Methods for calculating ordered signatures

1 I NormalSubgroupsForRep( groupdata, divisor ) O

Let groupdata be the output of RDS:PermutationRepForDiffsetCalculations and divisor an integer.
Then NormalSubgroupsForRep calculates all normal subgroups of groupdata.G such that the size of the
factor group is divisible by divisor and the factor group is a semidirect product of cyclic groups.

The output is a record consisting of

1. a normal subgroup .Nsg of G

2. the factor group .fgrp:=G/Nsg

3. the epimorphism .epi from G to .fgrp

4. a root of unity .root

5. a galois automorphism .alpha

6.+7. generators of the factor group G/.Nsg named .a and .b such that .a is normalized by .b.

8 a list .int2pairtable such that the i th entry is the pair [m,n] with that Glist[i]ˆepi=aˆ(m-1)*bˆ(n-1)

.alpha and .root may be used as input for RDS:OrderedSigs

2 I OrderedSigs( coeffSums, absSum, alpha, root ) O

Let G be group which contains a normal subgroup of index s such that the coset signature for a difference
set for this normal subgroup is coeffSums. Let N be a normal subgroup of G such that G/N is a semidirect
product of cyclic group of orders s, q and i divides the order of G/N .

Then OrderedSigs(coeffSums,absSum,alpha,root) calculates all ordered signatures for N . Here root is a
primitive q-th root of unity and alpha is a Galois- automorphism of CS (q) with order dividing s. absSum is
the order of the difference set. (i.e. order = k − λ).

OrderedSigs is based on calculations using an s-dimensional unitary representation of G/N . In this rep-
resentation a subset of G induces a semi-circular matrix. The returned value is a list of lists s-tuples The
entries of the s-tuples are coefficients of numbers in Z[root ] such that the semi-circular matrix defined by
these numbers together with alpha meets necessary conditions for matrices induced by difference sets. To
gain the algebraic numbers from the s-tuple tup, use List(tup,i->CoeffList2CyclotomicList(i,root))

Each |coeffSums|-tuple returned defines an ordered signature. The ordering of G/N is chosen to fit to the
data returned by RDS:NormalSubgroupsForRep:

[a0, a1, . . . , aq−1], [a0b, a1b, . . . , aq−1b], . . . , [a0bs−1, . . . , aq−1bs−1]

So for the calculation of ordered signatures, smaller ordered signatures coeffSums have to be known. But
this is not so bad, as small signatures are easy to calculate. The following example shows an application.

gap> G:=SmallGroup(273,3);
<pc group of size 273 with 3 generators>
gap> Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> CosetSignatures(273,273/3,16);
[ [ 3, 7, 7 ] ]
gap> nsgs:=NormalSubgroupsForRep(Gdata,3);
[ rec( Nsg := Group([ f2 ]), alpha := ANFAutomorphism( CF(13), 3 ),

root := E(13), fgrp := Group([ f1, <identity> of ..., f2 ]),
epi := [ f1, f2, f3 ] -> [ f1, <identity> of ..., f2 ], a := f2,
b := f1,
int2pairtable := [ [ 1, 1 ], [ 1, 2 ], [ 1, 1 ], [ 2, 1 ], [ 1, 3 ],

...
[ 8, 3 ], [ 11, 3 ], [ 5, 2 ], [ 11, 3 ] ] ),

rec( Nsg := Group([ f3 ]), alpha := ANFAutomorphism( CF(7), 2 ),
root := E(7), fgrp := Group([ f1, f2, <identity> of ... ]),
epi := [ f1, f2, f3 ] -> [ f1, f2, <identity> of ... ], a := f2,
b := f1,
int2pairtable := [ [ 1, 1 ], [ 1, 2 ], [ 2, 1 ], [ 1, 1 ], [ 1, 3 ],

...
[ 6, 3 ], [ 4, 3 ], [ 4, 2 ], [ 6, 3 ] ] ) ]

gap> osigs:=OrderedSigs([3,7,7],16,nsgs[2].alpha,nsgs[2].root);
[ [ [ 0, 0, 0, 1, 0, 1, 1 ], [ 0, 0, 1, 2, 2, 0, 2 ], [ 2, 2, 0, 2, 0, 0, 1 ] ],
[ [ 0, 0, 0, 1, 0, 1, 1 ], [ 0, 1, 2, 2, 0, 2, 0 ], [ 2, 0, 0, 1, 2, 2, 0 ] ],

...
[ [ 1, 1, 0, 1, 0, 0, 0 ], [ 2, 2, 1, 0, 0, 2, 0 ], [ 2, 1, 0, 0, 2, 0, 2 ] ] ]

gap> Size(osigs);
98
gap> Set(osigs,g->SortedList(Concatenation(g)));
[ [ 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 1, 1, 1, 1, 1, 2, 2, 2, 2, 2, 2 ] ]

Note that the signature [3, 7, 7] can be assumed to be ordered (by passing to a suitable translate). So
even if we are not interested in ordered signatures, we have found out that there is only one admissible
unordered signature for this normal subgroup. To get this result using RDS:TestedSignatures would have
taken a very long time.

Of course, ordered signatures can also be used directly.

3 I OrderedSignatureOfSet( set, normal data ) O

takes a set set of integers (meant to be a partial difference set) and a list of records as returned by
RDS:NormalSubgroupsForRep. The returned value is a list of lists which is the ordered signature of the
partial difference set set and can be compared to the output of RDS:OrderedSigs

gap> OrderedSignatureOfSet([2,3,4,5],nsgs[2]);
[ [ 1, 1, 1, 0, 0, 0, 0 ], [ 1, 0, 0, 0, 0, 0, 0 ], [ 0, 0, 0, 0, 0, 0, 0 ] ]



8
Block Designs and
Projective Planes

This section contains functions to help studying projective planes. There is also a function converting relative
difference sets to block designs. Those desings can be studied with the DESIGN [Soi06a] package by L. Soicher.

Projective planes are always assumed to consist of positive integers (as points) and sets of integers (as
blocks). The incidence relation is assumed to be the element relation. The blocks of a projective plane must
be sets.

1 I ProjectivePlane( blocks ) O

Given a list of lists blocks which represents the blocks of a projective plane, a block design is generated. If
the blocks is not a set of sets of the integers [1..v] for some v , the points are sorted and enumerated and
the blocks are changed accordingly. But the original names are known to the returned BlockDesign.

The block design generated this way will contain two extra entries, jblock and isProjectivePlane. The ma-
trix .jblock contains the number of the block containing the points i and j at the (i , j )th position. And
isProjectivePlane will be true. If blocks do not form the lines of a projective plane, an error is issued.

2 I PointJoiningLinesProjectivePlane( plane ) O

Returns a matrix which has as ij th entry the point wich is contained in the blocks with numbers i and j .
This matrix is also stored in plane. Some operations are faster if plane contains this matrix. If plane is not
a projective plane, an error is issued.

gap> b:=[ [ 1, 3 ], [ 1, 6 ], [ 2, 4 ], [ 2, 7 ],
> [ 3, 5 ], [ 4, 6 ], [ 5, 7 ] ];;
gap> plane:=ProjectivePlane(b);
rec( isBlockDesign := true, v := 7,

blocks := [ [ 1, 3 ], [ 1, 6 ], [ 2, 4 ], [ 2, 7 ],
[ 3, 5 ], [ 4, 6 ], [ 5, 7 ] ],

jblock := [ [ 0, 0, 1, 0, 0, 2, 0 ], [ 0, 0, 0, 3, 0, 0, 4 ],
[ 1, 0, 0, 0, 5, 0, 0 ], [ 0, 3, 0, 0, 0, 6, 0 ],
[ 0, 0, 5, 0, 0, 0, 7 ], [ 2, 0, 0, 6, 0, 0, 0 ],
[ 0, 4, 0, 0, 7, 0, 0 ] ],
isProjectivePlane := true )

gap> PointJoiningLinesProjectivePlane(plane);
[ [ 0, 1, 0, 0, 3, 0, 0 ], [ 1, 0, 0, 0, 0, 6, 0 ], [ 0, 0, 0, 2, 0, 4, 0 ],
[ 0, 0, 2, 0, 0, 0, 7 ], [ 3, 0, 0, 0, 0, 0, 5 ], [ 0, 6, 4, 0, 0, 0, 0 ],
[ 0, 0, 0, 7, 5, 0, 0 ] ]

gap> RecNames(plane);
[ "isBlockDesign", "v", "blocks", "jblock", "isProjectivePlane", "jpoint" ]

3 I DevelopmentOfRDS( diffset, Gdata ) O

This calculates the development of a (partial relative) difference set diffset in the group given by Gdata.
That is, the associated block design.

diffset can be given as a list of group elements or a list of integers (positions in the set of group elements).
Gdata can either be the record returned by RDS:PermutationRepForDiffsetCalculations or a group or
a set of group elements.

In either case, the returned object is a BlockDesign in the sense of L. Soichers DESIGN package.

gap> G:=CyclicGroup(21);; Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> AllDiffsets([2],[1..21],4,[],Gdata,1);
[ [ 2, 5, 16, 17 ], [ 2, 6, 10, 18 ] ]
gap> d1:=DevelopmentOfRDS(Set(G){[2,5,16,17]},Set(G));
rec( isBlockDesign := true, v := 21,
blocks := [ [ 1, 2, 5, 16, 17 ], [ 1, 3, 14, 15, 21 ], [ 1, 4, 8, 10, 13 ],

[ 1, 6, 7, 9, 20 ], [ 1, 11, 12, 18, 19 ], [ 2, 3, 9, 10, 12 ],
[ 2, 4, 7, 15, 19 ], [ 2, 6, 8, 11, 21 ], [ 2, 13, 14, 18, 20 ],
[ 3, 4, 6, 17, 18 ], [ 3, 5, 8, 19, 20 ], [ 3, 7, 11, 13, 16 ],
[ 4, 5, 9, 11, 14 ], [ 4, 12, 16, 20, 21 ], [ 5, 6, 12, 13, 15 ],
[ 5, 7, 10, 18, 21 ], [ 6, 10, 14, 16, 19 ], [ 7, 8, 12, 14, 17 ],
[ 8, 9, 15, 16, 18 ], [ 9, 13, 17, 19, 21 ], [ 10, 11, 15, 17, 20 ] ],

autSubgroup := <permutation group with 21 generators>,
pointNames := [ <identity> of ..., f1, f2, f1^2, f1*f2, f2^2, f1^2*f2,

f1*f2^2, f2^3, f1^2*f2^2, f1*f2^3, f2^4, f1^2*f2^3, f1*f2^4, f2^5,
f1^2*f2^4, f1*f2^5, f2^6, f1^2*f2^5, f1*f2^6, f1^2*f2^6 ],

blockSizes := [ 5 ], blockNumbers := [ 21 ], isSimple := true,
isBinary := true )

gap> d2:=DevelopmentOfRDS([2,5,16,17],Gdata);;
gap> d1=d2
true
gap> d1=DevelopmentOfRDS(Set(G){[2,5,16,17]},G);
true
gap> d1=DevelopmentOfRDS([2,5,16,17],G);
true

Note that equality for block designs means equality of records. So DevelopmentOfRDS generates exactly the
same record in each of the above examples. The output is in fact independent of the chosen data type of the
input (as long as it is valid). In particular, the design always knows its pointNames.

4 I ProjectiveClosureOfPointSet( points[, maxsize], plane ) O

Let plane be a projective plane. Let points be a set of non-collinear points (integers) of this plane. Then
ProjectiveClosureOfPointSet returns a record with the entries .closure and .embedding .

Here .closure is the projective closure of points (the smallest projectively closed subset of plane containing
the points points). It is not checked, whether this is a projective plane. As the BlockDesign .closure has
points [1..w] and plane has poins [1..v] with w ≤ v , we need an embedding of .closure into plane. This
embedding is the permutation .embedding . It is a permutation on [1..v] which takes the points of .closure
to a set of points in plane containing points and preserving incidence. Note that nothing is known about
the behaviour of .embedding on any point outside [1..w] and [1..w]^.embedding .

If maxsize is given and maxsize 6= 0, calculations are stopped if the closure is known to have at least maxsize
points and the plane plane is returned as .closure with the trivial permutation as embedding.

Let’s find a Baer subplane in the desarguesian plane of order 4:



Section 1. Isomorphisms and Collineations 39

gap> G:=CyclicGroup(21);; Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> AllDiffsets([2],[1..21],4,[],Gdata,1);
[ [ 2, 5, 16, 17 ], [ 2, 6, 10, 18 ] ]
gap> plane:=DevelopmentOfRDS([2,5,16,17],Gdata);;
gap> ProjectiveClosureOfPointSet([1,3,4],plane);
rec( closure := rec( isBlockDesign := true, v := 3,

blocks := [ [ 1, 2 ], [ 1, 3 ], [ 2, 3 ] ]
pointNames := [ <identity> of ..., f2, f1^2 ]),
embedding := (2,3,4) )

gap> IsProjectivePlane(last.closure);
false
gap> baer:=ProjectiveClosureOfPointSet([1,3,4,5],plane);;
gap> baer.closure.blocks;
[ [ 1, 2, 6 ], [ 1, 3, 5 ], [ 1, 4, 7 ], [ 2, 3, 7 ],
[ 2, 4, 5 ], [ 3, 4, 6 ], [ 5, 6, 7 ] ]

gap> IsProjectivePlane(baer.closure);
true
gap> Set(baer.closure.blocks,b->OnSets(b,baer.embedding));
[ [ 1, 3, 14 ], [ 1, 4, 8 ], [ 1, 5, 17 ], [ 3, 4, 17 ],
[ 3, 5, 8 ], [ 4, 5, 14 ], [ 8, 14, 17 ] ]

8.1 Isomorphisms and Collineations

Isomorphisms of projective planes are mappings which take points to points and blocks to blocks and respect
incidence. A collineation of a projective plane P is an isomorphism from P to P .

As projective planes are assumed to live on the integers, isomorphisms of projective planes are represented
by permutations. To test if a permutation on points is actually an isomorphism of projective planes, the
following methods can be used.

1 I IsIsomorphismOfProjectivePlanes( perm, plane1, plane2 ) O

Let plane1 , plane2 be two projective planes. IsIsomorphismOfProjectivePlanes test if the permutation
perm on points defines an isomorphism of the projective planes plane1 and plane2 .

2 I IsCollineationOfProjectivePlane( perm, plane ) O

Let plane be a projective plane and perm a permutation on the points of this plane. IsCollineationOf-
ProjectivePlane(perm,plane) returns true, if perm induces a collineation of plane.

This is just another form to call IsIsomorphismOfProjectivePlanes(perm,plane,plane)

3 I IsomorphismProjPlanesByGenerators( gens1, plane1, gens2, plane2 ) O
I IsomorphismProjPlanesByGeneratorsNC( gens1, plane1, gens2, plane2 ) O

Let gens1 be a list of points generating the projective plane plane1 and gens2 a list of generating points
for plane2 . Then a permutation is returned representing a mapping from the points of plane1 to those of
plane2 and taking the list gens1 to the list gens2 . If there is no such mapping which defines an isomorphism
of projective planes, fail is returned.

IsomorphismProjPlanesByGeneratorsNC does not check whether gens1 and gens2 really generate the
planes plane1 and plane2 .

Look at the above example again:

gap> P:=ProjectivePlane( [ [ 1, 2, 6 ], [ 1, 3, 5 ], [ 1, 4, 7 ],
> [ 2, 3, 7 ], [ 2, 4, 5 ], [ 3, 4, 6 ], [ 5, 6, 7 ] ]);;
gap> pi:=IsomorphismProjPlanesByGenerators([1,2,3,4],P,[2,4,6,7],P);
(1,2,4,7,3,6,5)
gap> IsIsomorphismOfProjectivePlanes(pi,P,P);
true
gap> IsCollineationOfProjectivePlane(pi,P);
true
gap> IsomorphismProjPlanesByGenerators([1,2,3,4],P,[1,2,3,5],P);
fail
gap> ProjectiveClosureOfPointSet([1,2,3,5],P).closure.v;
4

8.2 Central Collineations

Let φ be a collineation of a projective plane which fixes one point block-wise (the so-called centre) and one
block point-wise (the so-called axis). If the centre is contained in the axis, φ is called elation. Otherwise, φ
is called homology. The group of elations with given axis is called translation group of the plane (relative
to the chosen axis). A projective plane with transitive translation group is called translation plane. Here
transitivity is on the points outside the axis.

1 I ElationByPair( centre, axis, pair, plane ) O

Let centre be a point and axis a block of a projective plane plane . pair must be a pair of points outside axis
and lie on a block containing center . Then there is a unique collineation fixing axis pointwise and centre
blockwise (an elation) and taking point[1] to point[2] .

If one of the conditions is not met, an error is issued. This method is faster, if plane.jpoint is known (see
RDS:PointJoiningLinesProjectivePlane)

2 I AllElationsCentAx( centre, axis, plane[, "generators"] ) O

Let centre be a point and axis a block of the projective plane plane. AllElationsCentAx returns the group
of all elations with centre centre and axis axis as a group of permutations on the points of plane.

If “generators” is set, only a list of generators of the translation group is returned. This method is faster, if
plane.jpoint is known (see RDS:PointJoiningLinesProjectivePlane)

3 I AllElationsAx( axis, plane[, "generators"] ) O

Let axis be a block of a projective plane plane. AllElationsAx returns the group of all elations with axis
axis.

If “generators” is set, only a set of generators for the group of elations is returned. This method is faster, if
plane.jpoint is known (see RDS:PointJoiningLinesProjectivePlane)

gap> P:=ProjectivePlane( [ [ 1, 2, 6 ], [ 1, 3, 5 ], [ 1, 4, 7 ],
> [ 2, 3, 7 ], [ 2, 4, 5 ], [ 3, 4, 6 ], [ 5, 6, 7 ] ]);;
gap> pi:=ElationByPair(1,[1,2,6],[3,5],P);
(3,5)(4,7)
gap> AllElationsCentAx(1,[1,2,6],P);
Group([ (3,5)(4,7) ])
gap> AllElationsAx([1,2,6],P);
Group([ (3,5)(4,7), (3,7)(4,5) ])
gap> AllElationsAx([1,2,6],P);
Group([ (3,5)(4,7), (3,7)(4,5) ])
gap> Size(last);

4

4 I IsTranslationPlane( [infline, ]plane ) O

Returns true if the plane plane has a block b such that the group of elations with axis b is transitive outside
b.

If infline is given, only the group of elations with axis infline is considered. This is faster than calculating
the full translation group if the projective plane plane is not a translation plane. If plane is a translation
plane, the full translation group is calculated.

This method is faster, if plane.jpoint is known (see RDS:PointJoiningLinesProjectivePlane)

gap> AllElationsAx(P.blocks[1],P);
Group([ (3,5)(4,7), (3,7)(4,5) ])
gap> Size(last);
4
gap> IsTranslationPlane(P);
true

5 I HomologyByPair( centre, axis, pair, plane ) O

HomologyByPair returns the homology defined by the pair pair fixing centre blockwise and axis pointwise.
The returned permutation fixes axis pointwise and centre linewise and takes pair[1] to pair[2] .

6 I GroupOfHomologies( centre, axis, plane ) O

returns the group of homologies with centre centre and axis axis of the plane plane.

gap> HomologyByPair(3,[1,2,6],[4,5],P);
Error, The centre must be fixed blockwise called from
# ...
gap> GroupOfHomologies(3,[1,2,6],P);
Group(())

8.3 Collineations on Baer Subplanes

Let P be a projective plane of order n2. A subplane B of order n of P is called Baer subplane. Baer
suplanes are exactly the maximal subplanes of P .

1 I InducedCollineation( baerplane, baercoll, point, image, planedata, embedding ) O

If a projective plane contains a Baer subplane, collineations of the subplane may be lifted to the full plane.
If such an extension to the full plane exists, it is uniquely determined by the image of one point outside the
Baer plane.

Here baercoll is a collineation (a permutation of the points) of the projective plane baerplane. The permuta-
tion embedding is a permutation on the points of the full pane which converts the enumeration of baerplane
to that of the full plane. This means that the image of the points of baerplane under embedding is a subset
of the points of plane. Namely the one representing the Baer plane in the enumeration used for the whole
plane. point and image are points outside the Baer plane.

The data for baerplane and embedding can be calculated using RDS:ProjectiveClosureOfPointSet.

InducedCollineation returns a collineation of the full plane (as a permutation on the points of plane)
which takes point to image and acts on the Baer plane as baercoll does. If no such collineation exists, fail
is returned.

This method needs plane.jpoint . If it is unknown, it is calculated (see RDS:PointJoiningLinesProjectivePlane)

Let’s go back to an earlier example and find a planar collineation:

gap> G:=CyclicGroup(21);; Gdata:=PermutationRepForDiffsetCalculations(G);;
gap> AllDiffsets([2],[1..21],4,[],Gdata,1);
[ [ 2, 5, 16, 17 ], [ 2, 6, 10, 18 ] ]
gap> plane:=DevelopmentOfRDS([2,5,16,17],Gdata);;
gap> baer:=ProjectiveClosureOfPointSet([1,3,4,5],plane);;
gap> pi:=InducedCollineation(baer.closure,(),21,15,plane,baer.embedding);
(2,16)(6,18)(7,12)(9,11)(10,13)(15,21)(19,20)
gap> 21^pi;
15
gap> ForAll(OnSets([1..7],baer.embedding),i->i^pi=i);
true

8.4 Invariants for Projective Planes

The functions NrFanoPlanesAtPoints, PRank@RDS, FingerprintAntiFlag and FingerprintProjPlane cal-
culate invariants for finite projective planes. For more details see [Röd06] and [Moo95]. The values of some
of these invariants are available from the homepages of [Moo] and [Roy] for many planes.

1 I NrFanoPlanesAtPoints( points, plane ) O

For a projective plane plane, NrFanoPlanesAtPoints(points,plane) calculates the so-called Fano invariant.
That is, for each point in points, the number of subplanes of order 2 (so-called Fano planes) containing
this point is calculated. The method returns a list of pairs of the form [point ,number ] where number is the
number of Fano sub-planes in point .

This method is faster, if plane.jpoint is known (see RDS:PointJoiningLinesProjectivePlane). Indeed, if
plane.jpoint is not known, this method is very slow.

gap> G:=CyclicGroup(4^2+5);
<pc group of size 21 with 2 generators>
gap> diffset:=OneDiffset(G);
[ f1, f1*f2, f1^2*f2^4, f1*f2^5 ]
gap> P:=DevelopmentOfRDS(diffset,G);;
gap> NrFanoPlanesAtPoints([3],P);
[ [ 3, 240 ] ]

2 I IncidenceMatrix( plane ) O

returns a matrix I , where the columns are numbered by the blocks and the rows are numbered by points.
And I[i][j]=1 if and only if points[i] is incident (contained in) blocks[j] (an 0 else).

3 I PRank( plane, p ) O

Let I be the incidence matrix of the projective plane plane and p a prime power. The rank of I · I t as a
matrix over GF (p) is called p-rank of the projective plane. Here I t denotes the transposed matrix. Note
that this is a method within the RDS workspace, so it has to be called as PRank@RDS

gap> G:=CyclicGroup(2^2+3);
<pc group of size 7 with 1 generators>
gap> P:=DevelopmentOfRDS(OneDiffset(G),G);;
gap> IncidenceMatrix(P);
[ [ 1, 1, 1, 0, 0, 0, 0 ], [ 1, 0, 0, 1, 1, 0, 0 ], [ 0, 1, 0, 1, 0, 1, 0 ],
[ 1, 0, 0, 0, 0, 1, 1 ], [ 0, 0, 1, 1, 0, 0, 1 ], [ 0, 0, 1, 0, 1, 1, 0 ],
[ 0, 1, 0, 0, 1, 0, 1 ] ]

gap> PRank@RDS(P,3);
6

gap> PRank@RDS(P,2);
4

4 I FingerprintProjPlane( plane ) O

For each anti-flag (p, l) of a projective plane plane of order n, define an arbitrary but fixed enumeration of
the lines through p and the points on l . Say l1, . . . , ln+1 and p1, . . . , pn+1 The incidence relation defines a
canonical bijection between the li and the pi and hence a permutation on the indices 1, . . . ,n + 1. Let σ(p,l)
be this permutation.

Denote the points and lines of the plane by q1, . . . qn2+n+1 and e1, . . . , en2+n+1. Define the sign matrix as
Aij = sgn(σ(qi ,ej )) if (qi , ej ) is an anti-flag and = 0 if it is a flag. Then the fingerprint is defnied as the
multiset of the entries of |AAt |.

5 I FingerprintAntiFlag( point, linenr, plane ) O

Let m1, . . . ,mn+1 be the lines containing point and E1, . . . ,En+1 the points on the line given by linenr such
that Ei is incident with mi . Now label the points of mi as point = Pi,1, . . . ,Pi,n+1 = Ei and the lines of
Ei as line = l1, . . . , li,n+1 = mi . For i 6= j , each Pj ,k lies on exactly one line li,kσi,j containing Ei for some
permutation σi,j

Define a matrix A, where Ai,j is the sign of σi,j if i 6= j and Ai,i = 0 for all i . The partial fingerprint is the
multiset of entries of |AAt | where At denotes the transposed matrix of A.

Look at the above example again:

gap> NrFanoPlanesAtPoints([1,2,3],plane);
[ [ 1, 240 ], [ 2, 240 ], [ 3, 240 ] ]
gap> Set(NrFanoPlanesAtPoints([1..plane.v],plane),i->i[2])=[240];
true
gap> PRank@RDS(plane,2);
10
gap> PRank@RDS(plane,3);
21
gap> PRank@RDS(plane,5);
20
gap> FingerprintProjPlane(plane);
[ [ 12, 420 ], [ 16, 21 ] ]
gap> FingerprintAntiFlag(1,6,plane);
[ [ 3, 20 ], [ 4, 5 ] ]

Some functions

for everyday use

This chapter contains a number of functions that did not fit in anywhere else. Some of them might be useful
for other people, too, so they were included here.

9.1 Groups and actions

1 I OnSubgroups( subgroup, aut ) F

For a group G and an automorphism aut of G , OnSubgroups(subgroup,aut) is the image of subgroup under
aut

gap> G:=Group((1,2,3),(2,3));
Group([ (1,2,3), (2,3) ])
gap> alpha:=InnerAutomorphism(G,(1,2,3));
^(1,2,3)
gap> OnSubgroups(Subgroup(G,[(2,3)]),alpha);
Group([ (1,3) ])

2 I RepsCClassesGivenOrder( group, order ) O

RepsCClassesGivenOrder( group, order ) returns all elements of order order up to conjugacy. Note that
the representatives are not always the smallest elements of each conjugacy class.

gap> RepsCClassesGivenOrder(SymmetricGroup(5),2);
[ (4,5), (2,3)(4,5) ]

9.2 Iterators

1 I CartesianIterator( tuplelist ) O

Returns an iterator for Cartesian(tuplelist)

2 I ConcatenationOfIterators( iterlist ) F

ConcatenationOfIterators(iterlist) returns an iterator which runs through all iterators in iterlist . Note
that the returned iterator loops over the iterators in iterlist sequentially beginning with the first one.

gap> it:=Iterator([1,2,3]);;
gap> it2:=CartesianIterator([[9,10],[11]]);;
gap> cit:=ConcatenationOfIterators([it,it2]);;
gap> repeat
> Print(NextIterator(cit),",\c");
> until IsDoneIterator(cit);
1,2,3,[ 9, 11 ],[ 10, 11 ],



Section 3. Lists and Matrices 45

9.3 Lists and Matrices

1 I IsListOfIntegers( list ) P

IsListOfIntegers( list ) returns IsSubset(Integers, list ) if list is a dense list and false otherwise.

2 I List2Tuples( list, int ) O

If Size( list ) is divisible by int , List2Tuples( list,int) returns a list list2 of size int such that Concate-
nation( list2 )= list and every element of list2 has the same size.

gap> List2Tuples([1..6],2);
[ [ 1, 2, 3 ], [ 4, 5, 6 ] ]

3 I MatTimesTransMat( mat ) O

does the same as mat*TransposedMat( mat ) but uses slightly less space and time for large matrices.

4 I PartitionByFunctionNF( list, f ) O

PartitionByFunctionNF( list, f ) partitions the list list according to the values of the function f defined
on list . If f returns fail for some element of list , PartitionByFunctionNF( list, f ) enters a break loop.
Leaving the break loop with ’return;’ is safe because PartitionByFunctionNF treats fail as all other results
of f .

5 I PartitionByFunction( list, f ) O

PartitionByFunction( list, f ) partitions the list list according to the values of the function f defined
on list . All elements, for which f returns fail are omitted, so PartitionByFunction does not necessarily
return a partition. If InfoLevel(InfoRDS) is at least 2, the number of elements for which f returns fail is
shown (if fail is returned at all).

gap> PartitionByFunctionNF([-1..5],x->x^2);
[ [ 0 ], [ -1, 1 ], [ 2 ], [ 3 ], [ 4 ], [ 5 ] ]
gap> test:=function(x)
> if x>0 then return Sqrt(x);
> else return fail;
> fi;
> end;
function( x ) ... end
gap> PartitionByFunction([-1..5],test);
[ [ 1 ], [ 4 ], [ 5 ], [ 2 ], [ 3 ] ]
gap> SetInfoLevel(InfoRDS,2);
gap> PartitionByFunction([-1..5],test);
#I -2-
[ [ 1 ], [ 4 ], [ 5 ], [ 2 ], [ 3 ] ]
gap> PartitionByFunctionNF([-1..5],test);
Error, function returned <fail> called from
...
brk> return;
[ [ 1 ], [ 4 ], [ 5 ], [ 2 ], [ 3 ], [ -1, 0 ] ]

9.4 Cyclotomic numbers

1 I IsRootOfUnity( cyc ) P

IsRootOfUnity tests if a given cyclotomic is actually a root of unity.

2 I CoeffList2CyclotomicList( list, root ) O

CoeffList2CyclogomicList( list, root ) takes a list of integers list and a root of unity root and returns
a list list2 , where list2[i]=list[i]* rootˆ(i-1).

3 I AbssquareInCyclotomics( list, root ) O

For a list of integers and a root of unity, AbssquareInCyclotomics( list, root ) returns the modulus of
Sum(CoeffList2CyclotomicList( list, root )).

4 I CycsGivenCoeffSum( sum, root ) O

CycsGivenCoeffSum( sum, root ) returns all elements of Z[root ] such that the coefficient sum is sum
and all coefficients are non-negative. The returned list has the following form: The cyclotomic numbers
are represented by coefficients. RDS:CoeffList2CyclotomicList can be used to get the algebraic number
represented by list . The list is partitioned into equivalence classes of elements having the same modulus.
For each class the modulus is returned. This means that CycsGivenCoeffSum returns a list of pairs where
the first entry of each pair is the square of the modulus of an element of the second entry. And the second
entry is a list of coefficient lists of cyclotomics in Z[root ] having the coefficient sum sum.

gap> CycsGivenCoeffSum(3,E(3));
[ [ 0, [ [ 1, 1, 1 ] ] ],
[ 3, [ [ 0, 1, 2 ], [ 0, 2, 1 ], [ 1, 0, 2 ], [ 1, 2, 0 ], [ 2, 0, 1 ],

[ 2, 1, 0 ] ] ], [ 9, [ [ 0, 0, 3 ], [ 0, 3, 0 ], [ 3, 0, 0 ] ] ] ]
gap> CycsGivenCoeffSum(2,E(2));
[ [ 0, [ [ 1, 1 ] ] ], [ 4, [ [ 0, 2 ], [ 2, 0 ] ] ] ]

9.5 Filters and Categories

The following was originally posted at the GAP forum by Thomas Breuer [Bre05].

Each filter in GAP is either a simple filter or a meet of filters. For example, IsInt and IsPosRat are simple
filters, and IsPosInt is defined as their meet IsInt and IsPosRat.

Each simple filter is of one of the following kinds.

1. property: Such a filter is an operation, the filter value can be computed. The (unary) methods of this
operation must return true or false, and the return value is stored in the argument, except if the argument
is of a basic data type such as cyclotomic (including rationals and integers), finite field element, permuta-
tion, or internally represented list –the latter with a few exceptions. Examples of properties are IsFinite,
IsAbelian, IsSSortedList.

2. attribute tester: Such a filter is associated to an operation that has been created via DeclareAttribute, in
the sense that the value is true if and only if a return value for (a unary method of) this operation is stored
in the argument. Currently, attribute values are stored in objects in the filter IsAttributeStoringRep.
Examples of attribute testers are HasSize, HasCentre, HasDerivedSubgroup.

2.’ property tester: Such a filter is similar to an attribute tester, but the associated operation is a property. So
property testers can return true also if the argument is not in the filter IsAttributeStoringRep. Examples
of property testers are HasIsFinite, HasIsAbelian, HasIsSSortedList.

3. category or representation: These filters are not associated to operations, their values cannot be computed
but are set upon creation of an object and should not be changed later, such that for a filter of this kind,

one can rely on the fact that if the value is true then it was true already when the object in question was
created.

The distinction between representation and category is intended to express dependency on or independence of
the way how the object is stored internally. For example, IsPositionalObjectRep, IsComponentObjectRep,
and IsInternalRep are filters of the representation kind; the idea is that such filters are used in low level
methods, and that higher level methods can be implemented without referring to these filters.

Examples of categories are IsInt, IsRat, IsPerm, IsFFE, and filters expressing algebraic structures, such
as IsMagma, IsMagmaWithOne, IsAdditiveMagma. When one calls such a filter, one can be sure that no
computation is triggered. For example, whenever a quotient of two integers is formed, the result is clearly
in the filter IsRat, but the system also stores the value of IsInt, i.e., GAP does not support “unevaluated
rationals” for which the IsInt value is computed on demand and then stored.

4. other filters: Some filters do not belong to the above kinds, they are not associated to operations but they
are intended to be set (or even reset) by the user or by functions also after the creation of objects. Examples
are IsQuickPositionList, CanEasilyTestMembership, IsHandledByNiceBasis.

Each meet of filters can involve computable simple filters (properties, attribute and property testers) and
not computable simple filters (categories, representations, other filters). When one calls a meet of two filters
then the two filters from which the meet was formed are evaluated (if necessary). So a meet of filters is
computable only if at least one computable simple filter is involved.

1 I IsComputableFilter( filter ) F

’IsComputableFilter(filter)’ returns true if a the filter filter is computable. Filters for which ’IsComputable-
Filter’ returns false may be used in ’DeclareOperation’.

gap> IsComputableFilter( IsFinite );
true
gap> IsComputableFilter( HasSize );
true
gap> IsComputableFilter( HasIsFinite );
true
gap> IsComputableFilter( IsPositionalObjectRep );
false
gap> IsComputableFilter( IsInt );
false
gap> IsComputableFilter( IsQuickPositionList );
false
gap> IsComputableFilter( IsInt and IsPosRat );
false
gap> IsComputableFilter( IsMagma );
false

